
Introduccion

El objetivo del presente trabajo es presentar una coleccién de ejercicios para la asig-
natura “Programacion declarativa” de tercer curso de la Ingenieria Informatica.

Estos ejercicios complementa los apuntes de introduccién a la programacion decla-
rativa con Prolog ([]) y a las transparencias de clase ([2]).

Todos los ejercicios se han comprobado usando la versién 5.6.18 de SWI Prolog.

Capitulo 1

Operaciones con listas

Una lista es la lista vacia o se compone de un primer elemento y un resto, que es
una lista. En Prolog, la lista vacia se representa por [] y las listas no vacia son de la
forma [X|L] donde X es la cabeza y L es el resto.

1.1. Primer elemento

Ejercicio 1.1 Definir la relacién primero (7L, ?X) que se verifique si X es el primer elemento de
la lista L. Por ejemplo,

?- primero([a,b,c],X).
X=a

Obtener las respuestas a las siguientes preguntas:

7- primero([X,b,c],a).
7- primero([X,Y],a).
?7- primero(X,a).

Solucién: La definicién de primero es

primero([X[_]1,X).

Las respuestas a las preguntas son

7- primero([a,b,c],X).
X=a

7- primero([X,b,c],a).
X =a

?7- primero([X,Y],a).
X=a

Y=12Z
7- primero(X,a).
X = [alZ]

1.2. Resto de una lista

Ejercicio 1.2 Definir la relacion resto(7L1,?L2) que se verifique si L2 es la lista obtenida a
partir de la lista L1 suprimiendo el primer elemento. Por ejemplo,

?- resto([a,b,c],L).
L = [b, cl

Obtener las respuestas a las siguientes preguntas:

?- resto([alL], [b,c]).
?- resto(L,[b,c]).

Soluciéon: La definiciéon de resto es

resto([_|L],L).

Las respuestas a las preguntas son

?- resto([alL]l,[b,c]).
L = [b, c]

?- resto(L, [b,c]).

L =[X, b, c]

1.3. Construccion de listas

Ejercicio 1.3 Definir la relacion cons(?X,?L1,7L2) que se verifique si L2 es la lista obtenida
afiadiéndole X a L1 como primer elemento. Por ejemplo,

?- cons(a,[b,c],L).
L = [a, b, c]

Obtener las respuestas correspondientes a las siguientes preguntas:

?7- cons(X,[b,c],[a,b,c]).
?- cons(a,L,[a,b,c]).
?- cons(b,L,[a,b,c]).
?- cons(X,L,[a,b,c]).

Solucion: La definiciéon de cons es

cons (X,L, [XIL]).

Las respuestas a las preguntas son

?- cons(X, [b,c],[a,b,c]).
X=a

?- cons(a,L,[a,b,c]).

L = [b, c]

?- cons(b,L,[a,b,c]).

No

?- cons(X,L,[a,b,c]).

X =a

L = [b, c]

1.4. Relacién de pertenencia

Ejercicio 1.4 Definir la relacion pertenece(?X,?L) que se verifique si X es un elemento de la
lista L. Por ejemplo,

?7- pertenece(b, [a,b,c]).
Yes

7- pertenece(d, [a,b,c]).
No

Utilizar el programa para responder a las siguientes cuestiones:
1. ;Es c un elemento de [a,c,b,c]?
2. ;Cudles son los elementos de [a,b,a]?

3. ¢Cudles son los elementos comunes de [a,b,c]l, y [d,c,b]l?

Solucién: La definiciéon de pertenece(X,L), por recursiéon en L, es

pertenece(X, [XI_1).
pertenece(X, [_|L]) :-
pertenece(X,L).

Las respuesta a las preguntas son

1. ;Es c un elemento de [a,c,b,c]?

7- pertenece(c, [a,c,b,c]).
Yes

2. ;Cuales son los elementos de [a,b,al?

7- pertenece(X, [a,b,al).
X =a;

X=>b;

X =a;

No

3. ¢(Cuales son los elementos comunes de [a,b,c], y [d,c,b]?

7- pertenece(X,[a,b,c]), pertenece(X,[d,c,b]l).

X=D>b;
X =c¢c ;
No

Nota: La relacién pertenece se corresponde con la definida member.

1.5. Concatenacion de listas

Ejercicio 1.5 Definir la relacién conc (7L1,7L2,7L3) que se verifique si L3 es la lista obtenida
escribiendo los elementos de L2 a continuacion de los elementos de L1. Por ejemplo,

?- conc([a,b], [c,d,e]l,L).
L =1[a, b, ¢, d, €]

Utilizar el programa para responder a las siguientes cuestiones:
1. ;Qué lista hay que afiadirle al lista [a,b] para obtener [a,b,c,d]?
2. ;Qué listas hay que concatenar para obtener [a,bl?
3. ¢Pertenece b a la lista [a,b,c]?
4. ;Es [b,c] una sublista de [a,b,c,d]?
5. ¢Es [b,d] una sublista de [a,b,c,d]?

6. ;Cudl es el iltimo elemento de [b,a,c,d]?

Solucién: La definicién de conc(L1,L2,L3), por recursién en L1, es

conc([],L,L).
conc([X|L1],L2,[X|L3]) :-
conc(L1,L2,L3).

Las repuestas a las cuestiones son
1. ;Qué lista hay que afiadirle al lista [a,b] para obtener [a,b,c,d]?

?- conc([a,b],L,[a,b,c,d]).
L = [c, d]

2. ;Qué listas hay que concatenar para obtener [a,b]?

?- conc(L1,L2,[a,b]l).
L1 = []

L2 = [a, b] ;

L1 = [a]

L2 = [b] ;

L1 = [a, b]

L2 = [1 ;

No

3. ;Pertenece b a la lista [a,b,c]?

?- conc(L1,[blL2], [a,b,c]).
L1 = [a]

L2 = [c]

Yes

?7- conc(_,[bl_1,[a,b,c]).
Yes

4. ;Es [b,c] una sublista de [a,b,c,d]?

?- conc(_,[b,cl_1,[a,b,c,d]).
Yes

5. ;Es [b,d] una sublista de [a,b,c,d]?

?- conc(_, [b,d]|_]1,[a,b,c,d]).
No

6. ;Cudl es el dltimo elemento de [b,a,c,d]?

?- conc(_,[X],[b,a,c,d]).
X =4d

Nota: La relacién conc se corresponde con la definida append.

1.6. Lista inversa

Ejercicio 1.6 Definir la relacién inversa(+L1,-L2) que se verifique si L2 es la lista obtenida
invirtiendo el orden de los elementos de la lista L1. Por ejemplo,

?- inversa([a,b,c],L).
L = [c, b, a]

Solucién: Vamos a presentar dos definiciones de inversa(L1,L2). Ambas son por
recursion en L1.
Primera soluciéon: Usando la relaciéon append, se define inversa como

inversa_1([]1,[]).

inversa_1([X|L1],L2) :-
inversa_1(L1,L3),
append (L3, [X],L2).

Segunda solucién: Usando un acumulador, se define inversa como

inversa_2(L1,L2) :-
inversa_2_aux(L1,L2,[]).

La relacién inversa_2_aux(+L1,-L2,+L3) se verifica si L2 es la lista obtenida afia-
diendo la inversa de L1 a L3 y se define por recursién en L1 como sigue

inversa_2_aux([],L2,L2).
inversa_2_aux([X|L1],L2,L3) :-
inversa_2_aux(L1,L2,[X|L3]).

Nota: La relaciéon inversa se corresponde con la relaciéon definida reverse.

1.7. Palindromo

Ejercicio 1.7 Un palindromo es una palabra que se lee igual en los dos sentidos, por ejemplo
“0s0”. Definir la relacién palindromo (+L) que se verifique si la lista L es un palindromo. Por
ejemplo,

7- palindromo([o,s,0]).
Yes

7- palindromo([o,s,a]).
No

Solucidén: La definicién de palindromo es

palindromo(L) :-
reverse(L,L).

1.8. Ultimo elemento

Ejercicio 1.8 Definir la relacion dltimo (?X,?L) que se verifique si X es el iiltimo elemento de
la lista L. Por ejemplo,

?7- 1ltimo (X, [a,b,c,d]).
X=4d

?- 1ltimo(a,L).

L = [a] ;

L = [X, al] ;

L=1[X, Y, a]

Yes

Solucidén: Presentamos tres definiciones de iltimo.
Primera solucién: Usando append se define d1timo por

dltimo_1(X,L) :-
append(_, [X],L).

Segunda solucién: Usando reverse se define d41timo por

dltimo_2(X,L) :-
reverse(L, [X]_1).

Tercera solucién: Una definicion de dltimo(X,L) por recursién en L es

altimo_3(X, [X]).
altimo_3(X,[_IL]) :-
altimo_3(X,L).

Nota: La relacién altimo se corresponde con la relaciéon definida last.

1.9. Penultimo elemento

Ejercicio 1.9 Definir lar relacion pentltimo (7X,?L) que se verifique si X es el peniiltimo ele-
mento de la lista L. Por ejemplo,

?7- peniltimo(X, [a,b,c,d]l).
X=c

7- peniltimo(c,L).

L = [c, X] ;

L =[X, c, Y]

Yes

Solucién: Se presentan tres definiciones de pentiltimo.
Primera solucién: Usando append se define pentiltimo por

peniltimo_1(X,L) :-
append(_, [X,_],L).

Segunda solucién: Usando reverse se define peniltimo por

peniltimo_2(X,L) :-
reverse(L,[_,X|_1).

Tercera solucién: Una definicién de peniltimo(X,L) por recursién en L es

penaltimo_3(X, [X,_1).
pendltimo_3(X,[_,YIL]) :-
peniltimo_3(X, [YIL]).

1.10. Seleccion de un elemento

Ejercicio 1.10 Definir la relacion selecciona(?X,?L1,7L2) que se verifique si L2 es la lista
obtenida eliminando una ocurrencia de X en L1. Por ejemplo,

?- selecciona(a,[a,b,al,L).
L = [b, al] ;

L = [a, b] ;

No

?- selecciona(c,[a,b,al,L).
No

?- selecciona(a,L,[1,2]).




1.11. Inserciéon de un elemento en una lista 17

L=1[a, 1, 2] ;

L=1[1, a, 2] ;

L=1[1, 2, al] ;

No

?- selecciona(X,[1,2,3],[1,3]).
X=2;

No

Solucién: La definiciéon de selecciona(X,L1,L2), por recursién en L1, es

selecciona(X, [X|L],L).
selecciona(X, [Y|L1],[YIL2]) :-
selecciona(X,L1,L2).

Nota: La relacion selecciona se corresponde con la definida select.

1.11. Insercion de un elemento en una lista

Ejercicio 1.11 Definir la relacion inserta(?X,7L1,7L2) que se verifique si L2 es una lista
obtenida insertando X en L1. Por ejemplo,

?- inserta(a,[1,2],L).
L=1[a, 1, 2] ;
L=1[1, a, 2] ;
L=1[1, 2, al] ;

No

Solucidon: La definicién de inserta es

inserta(X,L1,L2) :-
select(X,L2,L1).

1.12. Sublista

Ejercicio 1.12 Definir la relacion sublista(?L1,?L2) que se verifique si L1 es una sublista de
L2. Por ejemplo,

?- sublista([b,c],[a,b,c,d]).
Yes

?- sublista([a,c],[a,b,c,d]).
No

L
L
L
Yes

[a, blX]

[X, a, blY]
[X, Y, a, blZ]

.
3

?- sublista([a,b],L).

.
3

Solucion: La definicién de sublista es

sublista(L1,L2)
append(_L3,L4,L2),
append(L1,_L5,L4).

1.13. Permutacion

Ejercicio 1.13 Definir la relacién permutacion(+L1,?L2) que se verifique si L2 es una per-

mutacion de L1. Por ejemplo,

[ e s e
1}

No

[a,

= [a,

[b,
[b,
[c,
[c,

c]
b]
cl
al
b]
al

7- permutacién([a,b,c],L).

Solucién: La definicién de permutacién(L1,L2), por recursién en L1 es

permutacién([], [1).
permutacion (L1, [X|L2])
select(X,L1,L3),
permutaci6én(L3,L2).

Nota: La relacion permutacién(L1,L2) es equivalente a la definida permutation(L2,L1).

1.14. Lista con todos sus elementos iguales

Ejercicio 1.14 Definir la relacion todos_iguales (+L) que se verifique si todos los elementos
de la lista L son iguales entre si. Por ejemplo,

7- todos_iguales([a,a,al).
Yes

?7- todos_iguales([a,b,al).
No

?7- todos_iguales([]).

Yes

Solucién: La definicién de todos_iguales es

todos_iguales([]).

todos_iguales([_]).

todos_iguales([X,X[L]) :-
todos_iguales([X|L]).

1.15. Paridad de la longitud de una lista

Ejercicio 1.15 Definir la relacion longitud_par (+L) que se verifique si la longitud de la lista
L es par. Por ejemplo,

?7- longitud_par([a,b]).

Yes

?- longitud_par([a,b,c]).

No

Solucién: La definicién de longitud_par, por recursion cruzada con la relacién

longitud_impar, es

longitud_par([]).
longitud_par([_|L]) :-
longitud_impar(L).

La relacién longitud_impar (+L) se verifica si la longitud de la lista L es impar. Por
ejemplo,
?- longitud_impar([a,b]).
No
?7- longitud_impar([a,b,c]).
Yes

La definicién de longitud_impar es

longitud_impar ([_]).
longitud_impar ([_|L]) :-
longitud_par(L).

1.16. Rotacion de un elemento

Ejercicio 1.16 Definir la relacion rota(?L1,7L2) que se verifique si L2 es la lista obtenida a
partir de L1 colocando su primer elemento al final. Por ejemplo,

?- rota([a,b,c,d],L).
L =1[b, ¢, d, al
?7- rota(L, [b,c,d,al).
L =1"[a, b, ¢, d]

Solucion: La definiciéon de rota es

rota([X|L1],L) :-
append (L1, [X],L).

1.17. Subconjunto

Ejercicio 1.17 Definir la relacion subconjunto (+L1,?L2) que se verifique si L2 es un subcon-
junto de L1. Por ejemplo,

?7- subconjunto([a,b,c,d], [b,d]).
Yes

7- subconjunto([a,b,c,d], [b,£f]).
No

?- subconjunto([a,b,c],L).

L = [a, b, c] ;

L = [a, b] ;

L = [a, c] ;

L = [a] ;

L = [b, c] ;

L = [b] ;

L = [c] ;

L=1[];

No

Solucién: La definicién de subconjunto(L1,L2), por recursién en L1, es

subconjunto([1,[1).

subconjunto ([X|L1], [X|L2]) :-
subconjunto(L1,L2).

subconjunto([_|L1],L2) :-
subconjunto(L1,L2).

Aritmética

2.1. Maximo de dos niumeros

Ejercicio 2.1 Definir la relacién maximo (+X,+Y,?Z) que se verifique si Z es el mdximo de X e
Y. Por ejemplo,

?7- maximo(2,3,X).

X =3
?7- maximo(3,2,X).
X =3

Soluciéon: La definicion de madximo es

maximo(X,Y,X) :-
X >=Y.

maximo(X,Y,Y) :-
X <Y.

Nota: En Prolog esta definida la funcién max(X,Y) que devuelve el maximo de X e
Y. Por ejemplo,

?7- X is max(5,10).
X =10

2.2. Factorial

Ejercicio 2.2 Definir la relacién factorial (+X,?Y) que se verifique si Y es el factorial de X.
Por ejemplo,

?- factorial(3,X).

X=26

21

Solucién: La definiciéon de factorial(X,Y), por recursion sobre X, es

factorial(i,1).
factorial(X,Y) :-
X>1,
X1 is X-1,
factorial(X1,Y1),
Y is X * Y1.

2.3. Sucesion de Fibonacci

Ejercicio 2.3 La sucesion de Fibonacci es 0,1,1,2,3,5,8,13,21,...en la que cada término, salvo
los dos primeros, es la suma de los dos anteriores. Definir la relacién £ibonacci (+N,-X) que se
verifique si X es el N—ésimo término de la sucesion de Fibonacci. Por ejemplo,

?- fibonacci(6,X).
X =28

Solucién: La definiciéon de fibonacci(N,X), por induccién en N, es

fibonacci(0,0).

fibonacci(1,1).

fibonacci(N,X) :-
N>1,
N1 is N-1,
fibonacci(N1,X1),
N2 is N-2,
fibonacci(N2,X2),
X is X1+X2.

2.4. Maximo comun divisor

Ejercicio 2.4 Definir la relacién mcd (+X,+Y,?Z) que se verifique si Z es el mdximo comiin
divisor de X e Y. Por ejemplo,

?7- mcd(10,15,X).
X=25

Solucion: La definiciéon de mcd es

mcd (X,X,X).
mcd(X,Y,Z) :-
X<Y,
Y1 is Y-X,
mcd(X,Y1,2).
mcd(X,Y,Z) :-
X >V,
mcd(Y,X,Z).

2.5. Longitud de una lista

Ejercicio 2.5 Definir la relacion longitud(?L,?N) que se verifique si N es la longitud de la
lista L. Por ejemplo

?7- longitud([a,b,c],N).
N=23

?- longitud(L,3).
L=[X, Y, Z]

Solucién: La definicién de longitud es

longitud([],0).
longitud([_|L],N) :-
longitud(L,N1),
N is N1 + 1.

Nota: La relaciéon longitud se corresponde con la relaciéon definida length.

2.6. Lista de numeros acotada por su longitud

Ejercicio 2.6 Una lista estd acotada si todos sus elementos son menores que su longitud. Definir
la relacion 1lista_acotada(+L) que se verifique si todos los elementos de la lista de niimeros L
son menores que la longitud de L. Por ejemplo,

?- lista_acotada([1,0,2]).
Yes
?- lista_acotada([1,3,2]).
No

Soluciéon: La definicién de lista_acotada es

lista_acotada(L) :-
length(L,N),
lista_acotada_aux(L,N).

donde lista_acotada_aux(+L,+N) se verifica si todos los elementos de la lista de nu-
meros L son menores que N. Por ejemplo,

?- lista_acotada_aux([1,5,3],7).
Yes

?- lista_acotada_aux([1,5,3],5).
No

y estd definida por

lista_acotada_aux([],_).

lista_acotada_aux([X|L],N) :-
X <N,
lista_acotada_aux(L,N).

2.7. Maximo de una lista de niumeros

Ejercicio 2.7 Definir la relacién max_lista(+L,?X) que se verifique si X es el mdximo de la
lista de miimeros L. Por ejemplo,

7- max_lista([1,3,9,5],X).
X=9

Solucidon: La definicién de max_lista es

max_lista([X],X).

max_lista([X1,X2[|L],Y) :-
X3 is max(X1,X2),
max_lista([X3|L],Y).

2.8. Suma de los elementos de una lista

Ejercicio 2.8 Definir la relacién suma_lista(+L,?X) que se verifique si X es la suma de los
elementos de la lista de niimeros L. Por ejemplo,

? - suma_lista([1,3,5],X).
X=29

Solucién: La definicion de suma_lista es

suma_lista([],0).
suma_lista([X|L],Y) :-
suma_lista(L, Y1),
Y is X+VY1.

Nota: La relacion suma_lista se corresponde con la relacion definida sumlist.

2.9. Lista de numeros ordenada

Ejercicio 2.9 Definir la relacion ordenada(+L) que se verifique si la lista de niimeros L estd
ordenada de manera creciente. Por ejemplo,

?- ordenada([1,3,3,5]).
Yes
?- ordenada([1,3,5,2]).
No

Solucidon: La definicidon de ordenada es

ordenada([_]).

ordenada([X,YIL]) :-
X =<Y,
ordenada([Y|L]).

2.10. Suma parcial de una lista

Ejercicio 2.10 Definir la relacion suma_parcial (+L1,+X,?L2) que se verifique si L2 es un
subconjunto de L1 tal que la suma de sus elementos es X. Por ejemplo,

7- suma_parcial([1,2,5,3,2],5,L).
L=1[1, 2, 2] ;

L =1[2, 3] ;

L = [5] ;

L = [3, 2] ;

No

Solucién: Se presentan dos definiciones de suma_parcial.
Primera solucién: Una definiciéon, usando subconjunto (p.20) y suma_lista (p.25),
es

suma_parcial_ 1(L1,X,L2) :-
subconjunto(L1,L2),
suma_lista(L2,X).

Segunda solucién: Una definicion recursiva de suma_parcial es

suma_parcial_2([],0,[]).
suma_parcial_ 2([X|L1],Y, [X|L2]) :-
Y >= X,
Z is Y-X,
suma_parcial_2(L1,Z,L2).
suma_parcial 2([_|L1],Y,L2) :-
suma_parcial_2(L1,Y,L2).

2.11. Lista de N veces el nimero N

Ejercicio 2.11 Definir la relacion 1ista(+N,-L) que se verifique si L es la lista de longitud N
cuyos elementos son N. Por ejemplo,

?- lista(3,L).
L = [3, 3, 3]

Solucidon: La definicidon de 1ista es

lista(N,L) :-
lista_aux(N,N,L).

donde lista_aux(+N,+M,-L) se verifica si L es la lista de longitud M cuyos elementos
son N. Por ejemplo,

?7- lista_aux(5,4,L).
L =[5, 5, 5, 5]

y se define por

lista_aux(_,0,[]).
lista_aux(N,M,[NIL]) :-
M >0,
M1 is M-1,
lista_aux(N,M1,L).

2.12. Generacion de lista de numeros

Ejercicio 2.12 Definir la relacion 1ista_de_nimeros (+N,+M,-L) que se verificasi L es la lista
de los nmiimeros desde N hasta M, ambos inclusive. Por ejemplo,

?- lista_de_nimeros(3,5,L).
L = [3, 4, 5]

?- lista_de_nimeros(3,2,L).
No

Soluciéon: La definicidon de lista_de_ntmeros es

lista_de_numeros(N,N, [N]).
lista_de_numeros(N,M, [N|L]) :-
N <M,
N1 is N+1,
lista_de_numeros(N1i,M,L).

Nota: La relacion lista_de_nimeros se corresponde con la definida numlist.

2.13. Intervalo entero

Ejercicio 2.13 Definir la relacion entre (+N1,+N2,7X) que se verifique si X es un niimero en-
tero tal que N1 < X < N2. Por ejemplo,

?7- entre(2,5,X).

X=2;
X=3;
X =4 ;
X =5
No

Solucion: La definicidon de entre es

entre(N1,N2,N1) :-
N1 =< N2.
entre(N1,N2,X) :-
N1 < N2,
N3 is Ni+1,
entre(N3,N2,X) .

Nota: La relacién entre se corresponde con la definida between.

2.14. K-ésimo elemento

Ejercicio 2.14 Definir la relacién elemento_en(+K,?L,?X) que se verifique si X es el K—ésimo
elemento de la lista L (se empieza a numerar en 1). Por ejemplo,

?- elemento_en(2,[a,b,c,d] ,X).
X=0D»

?7- elemento_en(2,L,b).

L = [X, bl Y]

Solucidon: La definicién de elemento_en es

elemento_en(1, [X]_],X).
elemento_en(X,[_|L],X) :-
K>1,
K1 is K-1,
elemento_en(K1,L,X).

Nota: La relacién elemento_en se corresponde con la relacion definida nthi.

2.15. Multiplicacién de las ocurrencias de los elementos
de una lista

Ejercicio 2.15 Definir la relacion multiplicada(+L1,+N,-L2) que se verifica si L2 es la lista
obtenida repitiendo N veces los elementos de la lista L1. Por ejemplo,

?7- multiplicada([a,b,c],3,L).
L=17[a, a, a, b, b, b, ¢, ¢, cl

Solucién: La definiciéon de multiplicada es

multiplicada(L1,N,L2) :-
multiplicada_aux(L1,N,N,L2).

donde multiplicada_aux(+L1,+K,+N,-L2) se verifica si L2 es la lista obtenida repitien-
do K veces el primer elemento de L1 y N veces los restantes elementos. Por ejemplo,

7- multiplicada_aux([a,b,c],2,3,L).
L=17[a, a, b, b, b, ¢, ¢, c]

Su definicién es

multiplicada_aux([],_,_,[1).
multiplicada_aux([_|L1],0,N,L2) :-
multiplicada_aux(L1,N,N,L2).
multiplicada_aux([X|L1],K,N, [XIL2]) :-
K> 0,
K1 is K-1,
multiplicada_aux([X|L1],K1,N,L2).

Capitulo 3

Estructuras

3.1. Segmentos como objetos estructurados

Ejercicio 3.1 Supongamos que representamos los puntos del plano mediante términos de la for-
ma
punto(X,Y)
donde X e Y son niimeros, y los segmentos del plano mediante términos de la forma
segmento (P1,P2)
donde P1 y P2 son los puntos extremos del segmento. Definir las relaciones
vertical(?S) y horizontal(?S)
que se verifiquen si el segmento S es vertical (resp. horizontal). Por ejemplo,

7- vertical(segmento(punto(1,2),punto(1,3))).
Yes

7- vertical(segmento(punto(1,2),punto(4,2))).
No

7- vertical(segmento(punto(1,2),punto(1,3))).
No

7- vertical(segmento(punto(1,2),punto(4,2))).
Yes

Usar el programa para responder a las siguientes cuestiones:
1. ;Es vertical el segmento que une los puntos (1,1) y (1,2)7?
2. ;Eswvertical el segmento que une los puntos (1,1) y (2,2)?
3. ;Hay alguin Y tal que el segmento que une los puntos (1,1) y (2,Y) sea vertical?
4. ;Hay algiin X tal que el segmento que une los puntos (1,2) y (X,3) sea vertical?

5. ¢Hay algiin Y tal que el segmento que une los puntos (1,1) y (2,Y) sea horizontal?

31

6. ¢Para qué puntos el segmento que comienza en (2,3) es vertical?

7. ¢Hay algiin segmento que sea horizontal y vertical?

Solucién: Las definiciones de vertical y horizontal son

vertical(seg(punto(X,_Y),punto(X,_Y1))).
horizontal(seg(punto(_X,Y),punto(_X1,Y))).

Las respuestas a las preguntas son
1. ;Es vertical el segmento que une los puntos (1,1) y (1,2)?

‘ ?7- vertical(seg(punto(1,1),punto(1,2))).

Yes
2. ;Es vertical el segmento que une los puntos (1,1) y (2,2)?
‘ 7- vertical(seg(punto(1,1),punto(2,2))).
No
3. ¢Hay algtn Y tal que el segmento que une los puntos (1,1) y (2,Y) sea vertical?

‘ 7- vertical(seg(punto(1,1),punto(2,Y))).
No

4. ;Hay algtn X tal que el segmento que une los puntos (1,2) y (X,3) sea vertical?

?7- vertical(seg(punto(1,2),punto(X,3))).

X =1
No
5. ¢(Hay algtin Y tal que el segmento que une los puntos (1,1) y (2,Y) sea horizontal?
?7- horizontal(seg(punto(1,1),punto(2,Y))).
Y =1,
No
6. ¢Para qué puntos el segmento que comienza en (2,3) es vertical?
7- vertical(seg(punto(2,3),P)).
P = punto(2, _G459) ;
No
7. ¢Hay algin segmento que sea horizontal y vertical?

?7- vertical(S),horizontal(S).

S = seg(punto(_G444, _G445), punto(_G444, _G445)) ;
No

?7- vertical(_) ,horizontal( ).

Yes

3.2. Base de datos familiar

Ejercicio 3.2 En este ejercicio vamos a trabajar con una base de datos familiar.
1. Representar la informacion relativa a las siguientes familias:

» En la primera familia,

o ¢l padre es Tomds Garcia Pérez, nacido el 7 de Mayo de 1960, trabaja de profesor
y gana 60 euros diarios;

o la madre es Ana Lopez Ruiz, nacida el 10 de marzo de 1962, trabaja de médica y
gana 90 euros diarios;

e ¢l hijo es Juan Garcia Lépez, nacido el 5 de Enero de 1980, estudiante;
e la hija es Maria Garcia Lépez, nacida el 12 de Abril de 1992, estudiante.

» En la sequnda familia,

e ¢l padre es José Pérez Ruiz, nacido el 6 de Marzo de 1963, trabaja de pintor y
gana 120 euros diarios;

o la madre es Luisa Gdlvez Pérez, nacida el 12 de Mayo de 1964, trabaja de médica
y gana 90 euros diarios;

e un hijo es Juan Luis Pérez Pérez, nacido el 5 de Febrero de 1990, estudiante;
e una hija es Maria José Pérez Pérez, nacida el 12 de Junio de 1992, estudiante;
e otro hijo es José Maria Pérez Pérez, nacido el 12 de Julio de 1994, estudiante.

2. Realizar las siguientes consultas:

m ,existe familia sin hijos?

m jexiste familia con un hijo?

» ;existe familia con dos hijos?
» ;existe familia con tres hijos?

w jexiste familia con cuatro hijos.?
3. Buscar los nombres de los padres de familia con tres hijos.

4. Definir la relacién casado (X) que se verifique si X es un hombre casado.

5. Preguntar por los hombres casados.
6. Definir la relacion casada(X) que se verifique si X es una mujer casada.
7. Preguntar por las mujeres casadas.
8. Determinar el nombre de todas las mujeres casadas que trabajan.
9. Definir la relacion hijo (X) que se verifique si X figura en alguna lista de hijos.
10. Preguntar por los hijos.
11. Definir la relacién persona(X) que se verifique si X es una persona existente en la base de
datos.
12. Preguntar por los nombres y apellidos de todas las personas existentes en la base de datos.
13. Determinar todos los estudiantes nacidos antes de 1993.
14. Definir la relacién fecha_de_nacimiento(X,Y) de forma que si X es una persona, enton-
ces Y es su fecha de nacimiento.
15. Buscar todos los hijos nacidos en 1992.
16. Definir la relacion sueldo (X,Y) que se verifique si el sueldo de la persona X es Y.
17. Buscar todas las personas nacidas antes de 1964 cuyo sueldo sea superior a 72 euros dia-
rios.
18. Definir la relacion total(L,Y) de forma que si L es una lista de personas, entonces Y es la
suma de los sueldos de las personas de la lista L.
19. Calcular los ingresos totales de cada familia.

Solucién: Solucién del apartado 1: La representacién de la informacién sobre las

dos familias es

familia(persona([tomas,garcia,perez],

fecha(7,mayo,1960),
trabajo(profesor,60)),
persona([ana,lopez,ruiz],
fecha(10,marzo,1962),
trabajo(medica,90)),
[ persona([juan,garcia,lopez],
fecha(5,enero,1990),
estudiante),

persona([maria,garcia,lopez],
fecha(12,abril, 1992),
estudiante) ]).

familia(persona([jose,perez,ruiz],
fecha(6,marzo,1963),
trabajo(pintor,120)),
persona([luisa,galvez,perez],
fecha(12,mayo,1964),
trabajo(medica,90)),

[ persona([juan_luis,perez,perez],
fecha(5,febrero,1990),
estudiante),

persona([maria_jose,perez,perez],
fecha(12, junio, 1992),
estudiante),

persona([jose_maria,perez,perez],
fecha(12,julio, 1994),
estudiante) 1).

Solucién del apartado 2: Las consultas, y sus respuestas son,

?- familia(_,_,[]).

No

?- familia(_,_,[_1).

No

?- familia(_,_,[_,_1).
Yes

?- familia(_,_,[_,_,_1).
Yes

?7- familia(_,_,[_,_,_,_1).
No

Solucién del apartado 3:

?- familia(persona(NP,_,_),_,[_,_,_1).
NP = [jose, perez, ruiz] ;
No

Solucién del apartado 4:

casado(X) :-
familia(X,_,_).

Solucién del apartado 5:

?7- casado(X).

X = persona([tomas, garcia, perez],
fecha(7, mayo, 1960),
trabajo(profesor, 60)) ;

X = persona([jose, perez, ruiz],
fecha(6, marzo, 1963),
trabajo(pintor, 120)) ;

No

Solucién del apartado 6:

casada(X) :-
familia(_,X,_).

Solucién del apartado 7:

7- casada(X).

X = persona([ana, lopez, ruiz],
fecha(10, marzo, 1962),
trabajo(medica, 90)) ;

X = persona([luisa, galvez, perez],
fecha(12, mayo, 1964),
trabajo(medica, 90)) ;

No

Solucién del apartado 8:

7- casada(persona([N,_,_1,_,trabajo(_,_))).
N = ana ;

N
No

luisa ;

Solucién del apartado 9:

hijo(X) :-
familia(_,_,L),
member (X,L) .

Solucién del apartado 10:

?7- hijo(X).

persona([juan,garcia,lopez] ,fecha(5,enero,1990),estudiante) ;
persona([maria,garcia,lopez],fecha(12,abril, 1992),estudiante) ;
persona([juan_luis,perez,perez],fecha(5,febrero,1990),estudiante)
persona([maria_jose,perez,perez],fecha(12,junio,1992),estudiante)
persona([jose_maria,perez,perez],fecha(12,julio,1994) ,estudiante)

Lo e B B
1}

No

Solucién del apartado 11:

.
3

persona(X) :-
casado(X);
casada(X);
hijo(X).

Solucién del apartado 12:

?- persona(persona(X,_,_)).
[tomas, garcia, perez] ;
= [jose, perez, ruiz] ;

= [ana, lopez, ruiz] ;

= [luisa, galvez, perez] ;
[juan, garcia, lopez] ;

= [maria, garcia, lopez] ;

= [juan_luis, perez, perez] ;
= [maria_jose, perez, perez] ;

Lo - - - - - -
1}

= [jose_maria, perez, perez] ;

=
o

Solucién del apartado 13:

7- persona(persona(X,fecha(_,_,Afio) ,estudiante)), Afio < 1993.
X = [juan, garcia, lopez]

Ao = 1990 ;

X = [maria, garcia, lopez]

Ao = 1992 ;

X = [juan_luis, perez, perez]

Afio = 1990 ;

X = [maria_jose, perez, perez]

Ao = 1992 ;

No

Solucién del apartado 14:



38 Capitulo 3. Estructuras

fecha_de_nacimiento(persona(_,Y,_),Y).

Solucién del apartado 15:

7- hijo(X),fecha_de_nacimiento(X,fecha(_,_,1992)).

X = persona([maria_jose,perez,perez],fecha(12,junio,1992),estudiante) ;
No

Solucién del apartado 16:

sueldo (persona(_,_,trabajo(_,Y)),Y).
sueldo(persona(_,_,estudiante),0).

Solucién del apartado 17:

7- persona(X),
fecha_de_nacimiento(X,fecha(_,_,Afio)),
Afio < 1964,
sueldo(X,Y),

Y > 72.

X = persona([jose, perez, ruiz],
fecha(6, marzo, 1963),
trabajo(pintor, 120))

Ao = 1963

Y = 120 ;

X = persona([ana, lopez, ruiz],
fecha(10, marzo, 1962),
trabajo(medica, 90))

Ao = 1962

Y = 90 ;

Solucién del apartado 18:

total([],0).
total ([XIL],Y) :-
sueldo(X,Y1),
total(L,Y2),
Y is Y1 + Y2.

Solucién del apartado 19:

?7- familia(X,Y,Z),total([X,Y|Z],Total).
X = persona([tomas,garcia,perez],

fecha(7,mayo,1960),

trabajo(profesor,60))
persona([ana,lopez,ruiz],

fecha(10,marzo,1962),

trabajo(medica,90))
[persona([juan,garcia,lopez],fecha(5,enero,1990),estudiante),
persona([maria,garcia,lopez],fecha(12,abril,1992),estudiante)]
Total = 150 ;
X

-
1}

N
1]

persona([jose,perez,ruiz],

fecha(6,marzo,1963),

trabajo(pintor,120))
persona([luisa,galvez,perez],

fecha(12,mayo,1964),

trabajo(medica,90))
[persona([juan_luis,perez,perez],fecha(5,febrero,1990),estudiante),
persona([maria_jose,perez,perez],fecha(12, junio,1982),estudiante)
persona([jose_maria,perez,perez],fecha(12, julio,1984),estudiante)]
Total = 210 ;
No

-
1}

N
1]

3.3. Automata no-determinista

Ejercicio 3.3 Consideremos el autémata representado por

siendo e3 el estado final.

1.

S ks W

Representar el automata utilizando las siguientes relaciones

» final(X) que se verifica si X es el estado final.

» trans(E1,X,E2) que se verifica si se puede pasar del estado E1 al estado E2 usando
la letra X.

= nulo(E1,E2) que se verifica si se puede pasar del estado E1 al estado E2 mediante un
movimiento nulo.

. Definir la relacién acepta(E,L) que se verifique si el autémata, a partir del estado E,

acepta la lista L. Por ejemplo,

7- acepta(el,[a,a,a,b]).
Yes

7- acepta(e2,[a,a,a,b]).
No

Determinar si el autémata acepta la lista [a,a,a,b].
Determinar los estados a partir de los cuales el autémata acepta la lista [a,b].
Determinar las palabras de longitud 3 aceptadas por el automata a partir del estado e1l.

Definir la relacion acepta_acotada_1(E,L,N) que se verifique si el autémata, a partir
del estado E, acepta la lista L y la longitud de L es N.

7. Buscar las cadenas aceptadas a partir de el con longitud 3.

8. Definir la relacién acepta_acotada_2(E,L,N) que se verifique si el autémata, a partir
del estado E, acepta la lista L y la longitud de L es menor o igual que N.

9. Buscar las cadenas aceptadas a partir de el con longitud menor o igual 3.

Solucién: Solucién del apartado 1:

final(e3).

trans(el,a,el).
trans(el,a,e2).
trans(el,b,el).
trans(e2,b,e3).
trans(e3,b,e4).

nulo(e2,e4).
nulo(e3,el).

Solucién del apartado 2:

acepta(E,[]) :-
final(E).

acepta(E, [X|L]) :-
trans(E,X,E1),
acepta(E1,L).

acepta(E,L) :-
nulo(E,E1),
acepta(E1,L).

Solucién del apartado 3:

?7- acepta(el,[a,a,a,bl).
Yes

Solucién del apartado 4:

7- acepta(E, [a,b]).
E=el ;

E=e3 ;

No

Solucién del apartado 5:

7- acepta(el, [X,Y,Z]).
= a

N < X N < <
1]
o M O O P

No

Solucién del apartado 6: Presentamos dos definiciones. La primera usando acepta

acepta_acotada_la(E,L,N) :-
length(L,N),
acepta(E,L).

La segunda definicion es una variaciéon de la definicién de acepta:

acepta_acotada_1b(E,[],0) :-
final(E).
acepta_acotada_1b(E, [X|L],N) :-
N >0,
trans(E,X,E1),
Mis N - 1,
acepta_acotada_1b(E1,L,M).
acepta_acotada_1b(E,L,N) :-
nulo(E,E1),
acepta_acotada_1b(E1,L,N).

Nota: La primera definiciéon es mds simple y eficiente que la segunda como se ob-
serva en el siguiente ejemplo

7- time(acepta_acotada_la(e2,_L,5000)).
% 10,026 inferences, 0.01 CPU in 0.01 seconds (126% CPU, 1002600 Lips)
7- time(acepta_acotada_1b(e2,_L,5000)).
% 20,035 inferences, 0.02 CPU in 0.02 seconds (1267 CPU, 1001750 Lips)

A partir de ahora, adoptaremos la definiciéon acepta_acotada_1a

acepta_acotada_1(E,L,M) :-
acepta_acotada_la(E,L,M).

Solucién del apartado 7:

7- acepta_acotada_1(el,L,3).
L =1[a, a, b] ;
L = [b, a, b] ;
No
Solucién del apartado 8: Presentamos dos definiciones. La primera usando acepta

acepta_acotada_2a(E,L,N) :-
between(0,N,M),
length(L,M),
acepta(E,L).

y la segunda modificando acepta

acepta_acotada_2b(E,[]1,_N) :-
final(E).
acepta_acotada_2b(E, [X|L],N) :-
N >0,
trans(E,X,E1),
M is N-1,
acepta_acotada_2b(E1,L,M).
acepta_acotada_2b(E,L,N) :-
N >0,
nulo(E,E1),
acepta_acotada_2b(E1,L,N).

Nota: La primera definicién es mds simple y eficiente que la segunda como se ob-
serva en el siguiente ejemplo
7- time(acepta_acotada_2a(el,_L,10000)).
% 47 inferences, 0.00 CPU in 0.00 seconds (0% CPU, Infinite Lips)
?7- time(acepta_acotada_2b(el,_L,10000)).
% 50,027 inferences, 0.07 CPU in 0.06 seconds (113} CPU, 714671 Lips)

A partir de ahora, adoptaremos la definiciéon acepta_acotada_2a

acepta_acotada_2(E,L,M) :-
acepta_acotada_2a(E,L,M).

Solucién del apartado 9:

7- acepta_acotada_2(el,L,3).
L = [a, a, b] ;

L = [a, b] ;

L [b, a, bl ;

No

3.4. El problema del mono y el pliatano

Ejercicio 3.4 Un mono se encuentra en la puerta de una habitacion. En el centro de la habitacion
hay un pldtano colgado del techo. EI mono estd hambriento y desea coger el pldtano, pero no lo
alcanza desde el suelo. En la ventana de la habitacion hay una silla que el mono puede usar. EI
mono puede realizar las siguientes acciones: pasear de un lugar a otro de la habitacion, empujar
la silla de un lugar a otro de la habitacion (si estd en el mismo lugar que la silla), subirse en la
silla (si estd en el mismo lugar que la silla) y coger el pldtano (si estd encima de la silla en el
centro de la habitacion).

Definir la relacién solucién(E,S) que se verifique si S es una sucesion de acciones que
aplicadas al estado E permiten al mono coger el pldtano. Por ejemplo,

7- solucidén(estado(puerta,suelo,ventana,sin),L).
L = [pasear(puerta,ventana),empujar(ventana,centro),subir,coger]

donde estado (PM,EM,PS,X) significa que el mono se encuentra en la posicién PM (puerta, centro
o0 ventana) encima de EM (suelo o silla), la silla se encuentra en la posicion PS (puerta, centro o
ventana) y el mono tiene (X = con) o no (X = sin) el pldtano.

Solucidn:

solucidén(estado(_,_,_,con),[]).

solucién(E1,[AIL]) :-
movimiento(E1,A,E2),
solucién(E2,L) .

La relaciéon movimiento (estado (PM1,EM1,PS1,X1),A,estado(PM2,EM2,PS2,X2)) se ve-
rifica si en el estado (PM1,EM1,PS1,X1) se puede aplicar la accién A y como resultado de
su aplicacion se pasa al estado (PM2,EM2,PS2,X2).

movimiento(estado(centro,silla,centro,sin),
coger,
estado(centro,silla,centro,con)).
movimiento(estado(X,suelo,X,U),
subir,
estado(X,silla,X,U)).
movimiento(estado(X1,suelo,X1,U),
empujar(X1,X2),
estado(X2,suelo,X2,0)).
movimiento(estado(X,suelo,Z,U),
pasear(X,Z),
estado(Z,suelo,Z,U)).

3.5. Movimientos del caballo del ajedrez

Ejercicio 3.5 Supongamos que los cuadros del tablero de ajedrez los representamos por pares de
niimeros [X,Y] con X e Y entre 1y 8.

1. Definir la relacion salta(+C1,7C2) que se verifica si el caballo puede pasar en un movi-
miento del cuadrado C1 al cuadrado C2. Por ejemplo,

?7- salta([1,1],9).
S=[3,2];

S=[2,3];

No

2. Definir la relacion camino (L) que se verifique si L es una lista de cuadrados representando
el camino recorrido por un caballo sobre un tablero vacio. Por ejemplo,

?7- camino([[1,1],C]).
c=[3,2];

c=[2,3];

No

3. Usando la relacion camino, escribir una pregunta para determinar los caminos de longitud
4 por los que puede desplazarse un caballo desde cuadro [2,1] hasta el otro extremo del
tablero (Y=8) de forma que en el segundo movimiento pase por el cuadro [5,4].

4. Calcular el menor nmiimero de movimientos necesarios para desplazar el caballo del cuadro
[1,1] al [2,2]. ;Cudntos caminos de dicha longitud hay de [1,1] a [2,2]?

Solucién: Solucién del apartado 1:

salta([X,Y], [X1,Y1]) :-
dxy (Dx,Dy),
X1 is X+Dx,
correcto(X1),
Y1 is Y+Dy,
correcto(Y1).

La relacion dxy (7X, ?Y) se verifica si un caballo puede moverse X espacios horizon-
tales e Y verticales.

dxy(2,1).
dxy(2,-1).
dxy(-2,1).
dxy(-2,-1).

dxy(1,2).
dxy(1,-2).
dxy(-1,2).
dxy(-1,-2).

La relacién correcto (+X) se verifica si X estd entre 1y 8.

correcto(X) :-
1 =< X,
X =< 8.

Solucién del apartado 2:

camino([_]).

camino([C1,C2|L]) :-
salta(C1,C2),
camino ([C2|L]).

Solucién del apartado 3:

?- camino([[2,1],C1,[5,4],C2, [X,8]1]1).
c1 = [4, 2] c2 = [6, 6] X=17;
C1 = [4, 2] c2 = [6, 6]
C1 = [4, 2] c2 = [4, 6]
Ci1 = [4, 2] Cc2 = [4, 6]
c1 = [3, 3] c2 = [6, 6]
c1 = [3, 3] c2 = [6, 6]
c1 = [3, 3] c2 = [4, 6]
c1 = [3, 3] c2 = [4, 6]
No

L T T - B I
1
w oo N W oo,

Solucién del apartado 4:

?- camino([[1,1],_,[2,211).

No

?- camino([[1,1],_,_,[2,21]1).

No

?- camino([[1,1],_,_,_,[2,2]11).

Yes

?- camino([[1,1],C2,C3,C4,[2,2]1]).

c2 = [3, 2] c3 = [5, 3] c4 = [3, 4] ;
c2 = [3, 2] c3 = [5, 3] c4 = [4, 1] ;

c2 = [3, 2] c3 = [5, 1] c4 = [4, 3] ;
c2 = [3, 2] c3 = [1, 3] c4 = [3, 4] ;
c2 = [3, 2] Cc3 = [2, 4] c4 = [4, 3] ;
c2 = [2, 3] Cc3 = [4, 2] C4 = [3, 4] ;
c2 = [2, 3] Cc3 = [3, 5] c4 = [1, 4] ;

c2 = [2, 3] c3 = [3, 5] c4 = [4, 3] ;
c2 = [2, 3] c3 = [3, 1] c4 = [4, 3] ;
c2 = [2, 3] c3 = [1, 5] c4 = [3, 4] ;
No

3.6. Maximo elemento de un arbol binario

Ejercicio 3.6 Un drbol binario es vacio o consta de tres partes: la raiz (que debe de ser un niimero
positivo), el subdrbol izquierdo (que debe ser un drbol binario) y el subdrbol derecho (que debe ser
un drbol binario). Usaremos la siguiente representacion

= nil representa el drbol vacio
= t(I,R,D) representa el drbol de la raiz R, subdrbol izquierdo I y subdrbol derecho D.

Por ejemplo, t (t(nil,2,nil),1,t(t(nil,4,nil),3,nil)) representa el drbol

Definir la relacion maximo (+T,-X) que se verifique si X es el mdximo de los nodos del drbol T.
Por ejemplo,

?7- maximo(nil,N).

N=20

?- maximo(t(nil,2,nil),N).

N =2

?7- maximo(t(t(nil,2,nil),3,nil) ,N).
N =3

Soluciéon: La definicion de madximo es

maximo(nil,0).
maximo(t(I,R,D),M):-
maximo (I,MI),

maximo(D,MD),
M1 is max(MI,MD),
M is max(R,M1).




Capitulo 4

Retroceso, corte y negacion

4.1. Ejemplos de uso del corte

Ejercicio 4.1 1. Definir la relacion £ (X,Y) de forma que:

m 51 X < 3, entonces Y = 0;
m 513 <X <6, entonces Y = 2;
m 516 <X, entonces Y = 4.

2. Construir el drbol de deduccién correspondiente a la cuestion
7- £(1,Y), 2 < Y.

3. Definir la relacion £_1(X,Y) a partir de la definicion de £ (X,Y), introduciendo un corte
al final de las dos primeras cldusulas.

4. Construir el drbol de deduccion correspondiente a la cuestion
7- £.1(1,Y), 2 < Y.

5. Construir el drbol de deduccion correspondiente a la cuestion
7- £_1(7,Y).

6. En el drbol anterior se observa que se efectiian comparaciones innecesarias (por ejemplo,
después de fallar la comparacion 7<3, efectiia la comparacion 3=<7). Definir la relacion
f_2(X,Y) suprimiendo en la definicion de £_1(X,Y) las comparaciones innecesarias.

7. Construir el drbol de deduccion correspondiente a la cuestion
7- £.2(7,Y).

49

8. Construir el drbol de deduccion correspondiente a la cuestion
7- £.2(1,Y), 2 < Y.

9. Definir la relacion £ _3(X,Y) a partir de la definicion de £_2(X,Y), suprimiendo los cortes.

10. Obtener las respuestas correspondientes a la cuestion
7- £.3(1,V).

Solucién: Solucién del apartado 1: La definicion de f es

f(X,0) :- X < 3.
f(X,2) :- 3 =X, X < 6.
f(X,4) :- 6 =< X.

Solucién del apartado 2: El arbol de deduccién se muestra en la figura Bl (pagina

BD).
f(1,Y), 2<Y
{X/1, Y/O} {XI1,Y/4}
{XI1,Y12}
3=<1, 1<6, 2<2 6=<1,2<4
Fallo Fallo
2<0
Fallo
Figura 4.1: Arbol de resolucion del apartado 2
Solucién del apartado 3: La definicién de £_1 es
f_1(X,0) :- X < 3, !.
f_1(X,2) :- 3 =<X, X<6, !
f_1(X,4) :- 6 =< X.
Solucién del apartado 4: El arbol de deduccion se muestra en la figura .2 (pagina
BD.
Solucién del apartado 5: El arbol de deduccioén se muestra en la figura .3 (pagina
BI.

Solucién del apartado 6: La definicién de £_2 es

f 1(LY), 2<Y

{X/1, YIO}

1<3, !, 2<0

I, 2<0

2<0
Falo

Figura 4.2: Arbol de resolucion del apartado 4

f_1(7,Y)
{XI7, Y0} {X17, Y14}
{XI7,Y12}
7<3,! 3=<7,7<6,! | 6=<7]
Fallo
] [
Fallo Exito

(Y14}

Figura 4.3: Arbol de resolucién del apartado 5

f_2(X,0) :- X <3, !.
f_2(X,2) :- X <6, !.
f_2(X,4).

Solucién del apartado 7: El arbol de deduccién se muestra en la figura B4l (pagina

B2).

f_1(7,Y)
{X/7,Y/0} {XI7,Y14}
{X17,Y12}
7<3,! 7<6, !
Fallo Fallo Exito

(Y14}

Figura 4.4: Arbol de resolucién del apartado 7

Solucién del apartado 8: El arbol de deduccion se muestra en la figura 1.3 (pagina

B2).

f 2(1Y), 2<Y

{X/1, YIO}

1<3, !, 2<0

I, 2<0

2<0
Falo

Figura 4.5: Arbol de resolucién del apartado 8

Solucién del apartado 9: La definicién de £_3 es

f_3(X,0) :- X < 3.
f_3(X,2) :- X < 6.
f_3(X,4).

Solucién del apartado 10: Las respuestas son

7- £.3(1,Y).
Y=0;
Y=2;

Y =4
No

.
3

4.2. Arboles de deduccién de memberchk

Ejercicio 4.2 La relacion memberchk estd definida por

memberchk (X, [X|_]) :- '.
memberchk (X, [_|L]) :- memberchk(X,L).

Escribir los drboles de SLD resolucion correspondientes a las siguientes preguntas
1. ?- memberchk(X,[a,b,c]), X=a.
2. ?- memberchk(X, [a,b,c]), X=b.
3. ?7- X=b, memberchk(X, [a,b,c]).

Solucidn:

4.3. Diferencia de conjuntos

Ejercicio 4.3 Definir la relacién diferencia(+C1,+C2,-C3) que se verifique si C3 es la dife-
rencia de los conjuntos C1 y C2. Por ejemplo,

?- diferencia([a,b],[b,c],X).
X = [a];
No

Definir una version con negacion (diferencia_1) y otra con corte (diferencia_2) y comparar
la eficiencia de las distintas definiciones con el ejemplo

?- numlist(1,1000,_L1), time(diferencia_1(_L1,_L1)).
?- numlist(1,1000,_L1), time(diferencia_2(_L1,_L1)).

memberchk(X,[a,b,c]), X=a. memberchk(X,[a,b,c]), X=Db. X=b, memberchk(X,[a,b,c]).
{Xla / {Xla / {X/b}
I, a=a I, a=b. memberchk(b,[a,b,c]).
a=a a=b memberchk(b,[b,c]).
- Fallo '/
Resp.: X=a '
L]
Resp.: X=b

Figura 4.6: Arbol de resoluciéon memberchk

Nota: La relacion diferencia se corresponde con la relacion definida subtract.

Solucién: 17 definicion (con negacion):

diferencia_1([],_,[1).

diferencia_1([X|L],L2,L3):-
member (X,L2),
diferencia_1(L,L2,L3).

diferencia_1([X|L],L2,[X|L3]):-
not (member (X,L2)),
diferencia_1(L,L2,L3).

22 definicién (con corte):

diferencia_2([1,_,[1).

diferencia_2([X|L],L2,L3):-
member (X,L2), !,
diferencia_2(L,L2,L3).

diferencia_2([X|L],L2, [X|L3]):-
% not (member(X,L2)),
diferencia_2(L,L2,L3).

3% definicién (con corte y memberchk):



4.4. Agregacion de un elemento a un conjunto 55

diferencia_3([],_,[]1).

diferencia_3([X|L],L2,L3):-
memberchk(X,L2), !,
diferencia_3(L,L2,L3).

diferencia_3([X|L],L2, [X|L3]):-
% not (memberchk (X,L2)),
diferencia_3(L,L2,L3).

Comparaciones:

?- numlist(1,1000,_L1), time(diferencia_1(_L1,_L1,[]1)).

% 501,501 inferences, 0,24 CPU in 0,25 seconds (97 CPU, 2089587 Lips)
?- numlist(1,1000,_L1), time(diferencia_2(_L1,_L1,[]1)).

% 501,501 inferences, 0,25 CPU in 0,29 seconds (87% CPU, 2006004 Lips)
?- numlist(1,1000,_L1), time(diferencia_3(_L1,_L1,[]1)).

% 2,001 inferences, 0,07 CPU in 0,07 seconds (96% CPU, 28586 Lips)

?- numlist(1,1000,_L1), time(subtract(_L1,_L1,[])).

% 2,001 inferences, 0,08 CPU in 0,08 seconds (105% CPU, 25012 Lips)

4.4. Agregacion de un elemento a un conjunto

Ejercicio 4.4 Definir larelacion agregar (+X,+L, -L1) que se verifique si L1 es la lista obtenida
afiadiéndole X a L, si X no pertenece a L y es L en caso contrario. Por ejemplo,

?7- agregar(a,[b,c],L).

L = [a,b,c]
?7- agregar(b, [b,c],L).
L = [b,c]

Hacer una versién con negacion y otra con corte.

Solucién: 17 definicién (con negacién):

agregar_1(X,L,L):-
member (X,L) .

agregar_1(X,L, [XIL]) :-
not (member (X,L)) .

22 definicién (con corte):

agregar_2(X,L,L):-
member (X,L), !.
agregar_2(X,L, [X|L]).

4.5. Separacién de una lista de niimeros en positivos y ne-
gativos

Ejercicio 4.5 Definir la relacién separa(+L1,+L2,-L3) que separa la lista de niimeros L1 en
dos listas: L2 formada por los niimeros positivos y L3 formada por los niimeros negativos o cero.
Por ejemplo,

?- separa([2,0,-3,5,0,2],L2,L3).
L2 = [2, 5, 2]

L3 [0, -3, 0]

Yes

Proponer dos soluciones, una sin corte y otra con corte.
Solucién: Definicién con negacion:

separa_1([],[]1,[1).
separa_1([N|RL1], [N|RL2],L3) :-
N >0,
separa_1(RL1,RL2,L3).
separa_1([N|RL1],L2,[N|RL3]) :-
N =<0,
separa_1(RL1,L2,RL3).

Definicién con corte:

separa_2([]1,[1,[1).

separa_2([N|RL1], [N|RL2],L3) :-
N>O0, !,
separa_2(RL1,RL2,L3).

separa_2([N|RL1],L2, [N|RL3]) :-
% N =<0,
separa_2(RL1,L2,RL3).

4.6. Suma de los nameros pares de una lista de ntimeros

Ejercicio 4.6 Definir la relacion suma_pares (+L,-N) que se verifica si N es la suma de todos
los miimeros pares de la lista de niimeros L. Por ejemplo,

7- suma_pares([2,3,4],N).

N=26

?7- suma_pares([1,3,5,6,9,11,24],N).
N = 30

Hacer una versién con negacion y otra con corte.

Solucién:
Version 1 (con negacion):

suma_pares_1([],0).
suma_pares_1([N|L],X) :-
par(N),
suma_pares_1(L,X1),
X is X1 + N.
suma_pares_1([_N|L],X) :-
not (par(_N)),
suma_pares_1(L,X).

par(N): -
N mod 2 =:= 0.

Version 2 (con corte):

suma_pares_2([1,0).
suma_pares_2([N|L],X) :-
par(N), !,
suma_pares_2(L,X1),
X is X1 + N.
suma_pares_2([_N|L],X) :-
% not(par(_N)),
suma_pares_2(L,X).

Versioén 3 (con corte y acumulador):

suma_pares_3(L,X) :-
suma_pares_3_aux(L,0,X).

suma_pares_3_aux([],Ac,Ac).
suma_pares_3_aux([N|L],Ac,X) :-
par(N), !,

Acl is Ac + N,
suma_pares_3_aux(L,Ac1,X).
suma_pares_3_aux([_N|L],Ac,X) :-

% not(par(_N)),
suma_pares_3_aux(L,Ac,X) .

4.7. Exponente de dos en la factorizacién de un niimero

Ejercicio 4.7 Definir la relacion exponente_de_dos (+N, -E) que se verifica si E es el exponen-
te de 2 en la descomposicién de N como producto de factores primos. Por ejemplo,

7- exponente_de_dos(40,E).
E=3
7- exponente_de_dos(49,E).
E=0

Hacer una versién con negacion y otra con corte.

Solucién: 17 Version (con negacion):

exponente_de_dos_1(N,E):-
N mod 2 =:= 0,
N1 is N / 2,
exponente_de_dos_1(N1,E1),
E is E1 + 1.
exponente_de_dos_1(N,0) :-
N mod 2 =\= 0.

22 Versioén (con corte):

exponente_de_dos_2(N,E) : -
N mod 2 =:= 0, !,
N1 is N / 2,
exponente_de_dos_2(N1,E1),
E is E1 + 1.
exponente_de_dos_2(_,0).

4.8. Transformacién de lista a conjunto

Ejercicio 4.8 Definir la relacion 1ista_a_conjunto (+L,-C) que se verifique si C es el conjun-
to correspondiente a la lista L (es decir, C contiene los mismos elementos que L en el mismo orden,
pero si L tiene elementos repetidos sélo se incluye en C la iiltima aparicion de cada elemento). Por
ejemplo,

7- lista_a_conjunto([b,a,b,d],C).
C = [a, b, d]

Nota: La relacion 1ista_a_conjunto se corresponde con la relacion definida 1ist_to_set.

Solucién: La definiciéon de 1ista_a_conjunto es

lista_a_conjunto([],[]).

lista_a_conjunto([XIL],C) :-
member (X,L),
lista_a_conjunto(L,C).

lista_a_conjunto([X|L], [XIC]) :-
\+ member (X,L),
lista_a_conjunto(L,C).

La definicién anterior puede simplificarse con cortes

lista_a_conjunto_1([]1,[]1).

lista_a_conjunto_1([X|L],C) :-
member (X,L), !,
lista_a_conjunto_1(L,C).

lista_a_conjunto_1([X|L], [XIC]) :-
% \+ member(X,L),
lista_a_conjunto_1(L,C).

3% definicién (con corte y memberchk):

lista_a_conjunto_3([],[]).

lista_a_conjunto_3([X|L],L2):-
memberchk(X,L), !,
lista_a_conjunto_3(L,L2).

lista_a_conjunto_3([X|L], [X|L2]):-
% not (member(X,L)),
lista_a_conjunto_3(L,L2).

Comparaciones:

?7- numlist(1,1000,_L1), time(lista_a_conjunto_1(_L1,_12)).

% 1,003,001 inferences, 0,40 CPU in 0,41 seconds (97} CPU, 2507503 Lips)
7- numlist(1,1000,_L1), time(lista_a_conjunto_2(_L1,_12)).

% 501,501 inferences, 0,26 CPU in 0,28 seconds (93% CPU, 1928850 Lips)
?7- numlist(1,1000,_L1), time(lista_a_conjunto_3(_L1,_12)).

% 3,001 inferences, 0,07 CPU in 0,08 seconds (90% CPU, 42871 Lips)

?7- numlist(1,1000,_L1), time(list_to_set(_L1,_12)).

% 3,004 inferences, 0,07 CPU in 0,08 seconds (93} CPU, 42914 Lips)

4.9. Signos de crecimientos de sucesiones numéricas

Ejercicio 4.9 Definir la relacién crecimientos(+L1,-L2) que se verifique si L2 es la lista
correspondientes a los crecimientos de la lista numérica L1; es decir, entre cada par de elementos
consecutivos X e Y de L1 coloca el signo +si X < Y ey signo - en caso contrario. Por ejemplo,

?- crecimientos([1,3,2,2,5,3],L).
L = [13 +s 33 ) 2, T 2, +s 53 _]

Dar una definicién sin corte y otra con corte.

Solucidon: La definicidon de crecimientos sin usar corte es

crecimientos_1([_],[1).
crecimientos_1([X,Y|L1],[X,+|L2]) :-
X <Y,
crecimientos_1([Y|L1],L2).
crecimientos_1([X,Y|L1], [X,-I1L2]) :-
X>=Y,
crecimientos_1([Y|L1],L2).

La definicion de crecimientos usando corte es

crecimientos_2([_],[1).

crecimientos_2([X,Y|L1],[X,+|L2]) :-
X<y, o,
crecimientos_2([Y|L1],L2).

crecimientos_2([X,Y|L1],[X,-|L2]) :-
%X >= Y,
crecimientos_2([Y|L1],L2).

4.10. Descomposicién en factores primos

Ejercicio 4.10 Definir las siguientes relaciones:

= menor_divisor_propio(+N,?X) que se verifique si X es el menor divisor de N mayor o
igual que 2. Por ejemplo,

?7- menor_divisor_propio(30,X).
X =2
?7- menor_divisor_propio(3,X).
X =3

m factorizacidén(+N,-L) que se verifique si L es la lista correspondiente a la descomposi-
cion del niimero N en factores primos (se considera los que elementos de L estdn ordenados
de manera creciente). Por ejemplo,

?- factorizacidén(12,L).
L=1[2, 2, 3];

No

?- factorizacidén(1,L).
L=1[;

No

Solucién: La definicién de menor_divisor_propio es

menor_divisor_propio(N,X) :-
N1 is floor(sqrt(N)),
between(2,N1,X),
N mod X =:= 0, !.
menor_divisor_propio(N,N).

La definiciéon de factorizacidn es

factorizacioén(1,[]).

factorizacién(N, [XIL]) :-
N>1,
menor_divisor_propio(N,X),
N1 is N/X,
factorizacién(N1,L).

4.11. Menor elemento que cumple una propiedad

Ejercicio 4.11 Definir la relacion calcula(+N,+M,?X) que se verifique si X es el menor muilti-
plo de N tal que la suma de sus digitos es mayor que M. Por ejemplo,

?- calcula(3,10,X).
X = 39

Yes

?- calcula(7,20,X).
X = 399

Yes

Solucidon: La definicién de calcula es

calcula(N,M,X) :-
maltiplo(N,X),
suma_digitos(X,N1),
N1 > M, !.

La relacién maltiplo (+N,-X) se verifica si X es un multiplo de N. Por ejemplo,

7- miltiplo(5,X).
X =5

X =10 ;

X =15

Yes

miltiplo(N,N).

miltiplo(N,M) :-
maltiplo(N,N1),
M is N+N1.

La relaciéon suma_digitos(+N,-S) se verifica si S es la suma de los digitos del nt-
mero N. Por ejemplo,

?7- suma_digitos(237,9).
S =12

suma_digitos(N,N) :-
N < 10, !.
suma_digitos(N,S) :-
% N >= 10,
N1 is N // 10,
R is N - 10%*N1i,
suma_digitos(N1,S1),
S is S1 + R.

4.12. Numeros libres de cuadrados

Ejercicio 4.12 Un niimero es libre de cuadrados si no es divisible por el cuadrado de ningiin nii-
mero mayor que 1. Definir la relacion 1ibre_de_cuadrados (+N) que se verifique si el niimero
N es libre de cuadrados. Por ejemplo,

?7- libre_de_cuadrados(30).
Yes

?7- libre_de_cuadrados(12).
No

Solucidon: La definicidon de 1ibre_de_cuadrados es

libre_de_cuadrados(N) :-
M is floor(sqrt(N)),
not ((between(2,M,X), N mod (X*X) =:= 0)).

4.13. Suma de los numeros libres de cuadrados

Ejercicio 4.13 Definir la relacién suma_libres_de_cuadrados (+L,-S) que se verifique si S
es la suma de los niimeros libres de cuadrados la lista numérica L. Por ejemplo,

?- suma_libres_de_cuadrados([6,12,18,30],9).
S = 36

Nota: Dar dos definiciones, una con negacion y otra con corte.

Solucién: La definicién de suma_libres_de_cuadrados usando la negacién es

suma_libres_de_cuadrados_1([],0).
suma_libres_de_cuadrados_1([X|L],S) :-
libre_de_cuadrados(X),
suma_libres_de_cuadrados_1(L,S1),
S is X+S1.
suma_libres_de_cuadrados_1([X|L],S) :-
not (libre_de_cuadrados (X)),
suma_libres_de_cuadrados_1(L,S).

y la definicién usando corte es

suma_libres_de_cuadrados_2([]1,0).
suma_libres_de_cuadrados_2([X|L],S) :-
libre_de_cuadrados(X), !,
suma_libres_de_cuadrados_2(L,S1),
S is X+S1.
suma_libres_de_cuadrados_2([_X|L],S) :-
% not(libre_de_cuadrados(_X)),
suma_libres_de_cuadrados_2(L,S).

4.14. MaAaximo numero de una lista

Ejercicio 4.14 Definir la relacién max_lista(+L,-N) que se verifique si N es el mayor niimero
de la lista L. Por ejemplo,

?- max_lista([2,a23,5,7+9],N).

N =25

?7- max_lista([-2,a23,-5,7+9],N).
N = -2

?7- max_lista([a23,7+9],N).

No

Solucion: La definicion de max_lista es

max_lista(L,M) :-
member (M, L),
number (M) ,
not ((member (N,L),
number () ,
N>M)) .

4.15. Longitud de las subsucesiones comunes maximales

Ejercicio 4.15 Definir la relacion longitud_scm(+L1,+L2,-N) que se verifique si N es la lon-
gitud de las subsucesiones comunes maximales de las listas L1 y L2. Por ejemplo,

?7- longitud_scm([2,1,4,5,2,3,5,2,4,3],[1,7,5,3,2],N).
N =4,
No

ya que [1,5,3,2] es una subsucesion de las dos listas y no poseen ninguna otra subsucesion
comiin de mayor longitud. Obsérvese que los elementos de la subsucesion no son necesariamente
elementos adyacentes en las listas.

Solucién: La definicién de longitud_scm es

longitud_scm([],_,0).
longitud_scm(_, [1,0).
longitud_scm([X|L1], [X|L2],N) :-
!, longitud_scm(L1,L2,M),
N is M+1.
longitud_scm([X|L1], [Y|L2],N) :-

hX\=1Y,

longitud_scm(L1, [Y|L2],N1),
longitud_scm([X|L1],L2,N2),
N is max(N1,N2).

4.16. Elementos repetidos en una lista

Ejercicio 4.16 Definir la relacion repetido (-A,+L) que se verifique si el elemento A estd repe-
tido (i.e. ocurre mds de una vez) en la lista L. Por ejemplo,

?7- repetido(A,[1,2,1,3,4,3]).
A=1;

A=1;

A =3 ;

A =3
No

7- repetido(A,[1,2,5]).
No

3

Solucidén: La definicién de repetido es

repetido(A,L) :-
select(A,L,R),
memberchk (A,R) .

Ejercicio 4.17 Defimir la relacién elimina (+X,+L1,-L2) que se verifique si L2 es la lista ob-
tenida eliminando todas las ocurrencias de X en la lista L1. Por ejemplo,

?- elimina(a,[1,a,b,3,a,a,4,a,c],L).
L=1[1,0b, 3, 4, c]

Solucidon: La definicién de elimina es

elimina(_,[1,[]1).

elimina(X, [X|L1],L2) :-
elimina(X,L1,L2).

elimina(X,[YIL1],[YIL2]) :-
X\=1Y,
elimina(X,L1,L2).

Ejercicio 4.18 Definir la relacion repetidos (+L1,-L2) que se verifique si L2 es la lista de los
elementos repetidos de la lista L1. Por ejemplo,

?- repetidos_l([1,2,4,3,4,1,3,5],L).
L =1[1, 4, 3]

Solucién: La primera definicién de repetidos es

repetidos_1([1,[1).
repetidos_1([XIL1], [XIL2]) :-
memberchk (X,L1),
elimina(X,L1,L3),
repetidos_1(L3,L2).
repetidos_1([X|L1],L2) :-
not (memberchk (X,L1)),
repetidos_1(L1,L2).

con cortes se transforma en

repetidos_2([]1,[]1).
repetidos_2([X|L1], [X|L2]) :-
memberchk(X,L1), !,
elimina(X,L1,L3),
repetidos_2(L3,L2).
repetidos_2([_X|L1],L2) :-
% not (memberchk (_X,L1)),
repetidos_2(L1,L2).

4.17. Subconjunto maximal

Ejercicio 4.19 Definir la relacién subconjunto_maximal(+L1,-L2) que se verifica si L2 es
un subconjunto maximal de L1 (es decir, es un conjunto de elementos de L1 tal que sélo existe
un elemento de L1 que no pertenece a L2). Por ejemplo,

7- subconjunto_maximal([c,b,a,b,c,a,c],L).

L =[b, c] ;
L = [a, c] ;
L = [a, b] ;
No

Solucién: La definiciéon de subconjunto_maximal es

subconjunto_maximal (L1,L2) :-
list_to_set(L1,L3),
select(_,L3,L2).

Nota: La relacién list_to_set se corresponde con la relaciéon lista_a_conjunto
definida en la pagina

4.18. Suma de los elementos con posiciones miltiplos de
n

Ejercicio 4.20 Definir la relacion suma_posiciones(+N,+L,-S) que se verifique si S es la
suma de los elementos de la lista que ocupan las posiciones que son miiltiplos de N. Por ejemplo,

?7-suma_posiciones(2,[3,5,7,9,1,2],5).

S = 16
7- suma_posiciones(3,[3,5,7,9,1,2],S).
S =9

Solucién: La definiciéon de suma_posiciones es

suma_posiciones(N,L,S) :-
elemento_y_resto(N,L,X,L1), !,
suma_posiciones(N,L1,S1),
S is X+S1.
suma_posiciones(_,_,0).

donde elemento_y_resto(+N,+L1,-X,-L2) se verifica si X es el elemento N—ésimo de L1
y L2 es la lista L1 a partir del elemento X. Por ejemplo,

?7- elemento_y_resto(3,[3,5,7,9,1,2],X,L).
X =17
L=1[9, 1, 2]

La definiciéon de elemento_y_resto es

elemento_y_resto(N,L1,X,L2) :-
length(L,N),
append(L,L2,L1),
last(L,X).

4.19. Compresion de listas

Ejercicio 4.21 Definir la relacion comprimida (+L1,-L2) que se verifique si L2 es la lista obte-
nida sustituyendo cada sucesion de un elemento de L1 por dicho elemento. Por ejemplo,

?7- comprimida([a,b,b,a,a,a,c,c,b,b,b],L).
L =1[a, b, a, ¢, b]

Solucién: Vamos a presentar dos definiciones. La primera definicién de comprimida
es

comprimida([1,[]).
comprimida([X], [X]).
comprimida([X,X|L1],L2) :-
comprimida([X|L1],L2).
comprimida([X,Y|L1], [X|L2]) :-
X \=1Y,
comprimida([Y|L1],L2).

En la segunda definicién se usa el corte

comprimida_2([]1,[1).
comprimida_2([X1, [X]).
comprimida_2([X,Y|L1],L2) :-
X=Y, !,
comprimida_2([X|L1],L2).
comprimida_2([X,Y|L1], [X|L2]) :-
hX\=1Y,
comprimida_2([Y|L1],L2).

4.20. Empaquetamiento de listas

Ejercicio 4.22 Definir la relacién empaquetada(+L1,-L2) que se verifique si L2 es la lista
obtenida sustituyendo cada sucesion de un elemento de L1 por la lista formada por dicha sucesion.
Por ejemplo,

?7- empaquetada([a,b,b,a,a,a,c,c,b,b,b],L).
L = [[al, [b, b], [a, a, al, [c, c], [b, b, bl]

Solucién: La definicién de empaquetada es

empaquetada([], [1).

empaquetada([X|L1], [L2|L3]) :-
empaquetada_aux(X,L1,L4,L2),
empaquetada(L4,L3).

La relacion empaquetada_aux(X,L1,L4,L2) se verifica si L4 es la lista obtenida eli-
minando en L1 todas las ocurrencias iniciales de X y L2 es la lista formada por X y las
ocurrencias iniciales de X en L1; por ejemplo,

?7- empaquetada_aux(a,[a,a,c,c,b,b,b],L4,L2).
L4 = [C, C, b, b, b:]
L2 [a, a, a]

La definicién de empaquetada_aux es

empaquetada_aux (X, [1, [1, [X]).

empaquetada_aux (X, [X|L1],L4, [X|L2]) :-
empaquetada_aux(X,L1,L4,L2).

empaquetada_aux (X, [Y|L1], [YIL1],[X]) :-
X \=1Y.

La definicién anterior puede transformarse introduciendo corte en

empaquetada_aux_2(X, [1, [, [X]).

empaquetada_aux_2 (X, [X|L1],L4, [X|L2]) :-
I

‘)

empaquetada_aux_2(X,L1,L4,L2).
empaquetada_aux_2(X, [Y|L1],[YIL1], [X]).

4.21. Codificacion por longitud

Ejercicio 4.23 Definir la relacién codificada(+L1,-L2) que se verifique si L2 es la codifica-
cion por longitud de la lista L1; es decir, las sucesiones de un mismo elemento X de L1 se codifican
por términos de la forma N-X donde N es la longitud de la sucesion. Por ejemplo,

?- codificada([a,b,b,a,a,a,c,c,b,b,b],L).
L = [1-a, 2-b, 3-a, 2-c, 3-b]

(Indicacion: Usar la relacion empaquetada B2Z0)).

Soluciéon: La definiciéon de codificada es

codificada(L1,L2) :-
empaquetada(L1,L),
codificada_aux(L,L2).

Larelacién codificada_aux(+L1,-L2) se verifica si, suponiendo que L1 es una lista
de la forma [[E1,...,E1],...,[Em,....Em]]), L2 es la lista [N1-E1,...Nm-Em] donde
Ni es la longitud de [Ei, .. .,Ei]. Por ejemplo.

?- codificada_aux([[a],[b,b],[a,a,a],[c,c],[b,b,b]],L).
L = [1-a, 2-b, 3-a, 2-c, 3-b]

La definiciéon de codificada_aux es

codificada_aux([]1,[1).

codificada_aux([[X|Y]IL1], [N-X|L2]) :-
length([X|Y],N),
codificada_aux(L1,L2).

4.22. Codificacion reducida por longitud

Ejercicio 4.24 Definir la relacion codificada_reducida(+L1,-L2) que se verifique si L2 es
la codificacion reducida por longitud de la lista L1; es decir, las sucesiones de un mismo elemento
X de L1 se codifican por términos de la forma N-X donde N es la longitud de la sucesion cuando N
es mayor que 1y por X cuando N es igual a 1. Por ejemplo,

?- codificada_reducida([a,b,b,a,a,a,c,b,b,b],L).
L = [a, 2-b, 3-a, ¢, 3-b]

(Indicacion: Usar la relacién codificada @2T).

Soluciéon: La definiciéon de codificada_reducida es

codificada_reducida(L1,L2) :-
codificada(L1,L),
codificada_reducida_aux(L,L2).

La relaciéon codificada_reducida_aux(+L1,-L2) se verifica si L2 es la lista obtenida
transformando los elementos de L1 de la forma 1-X por X y dejando los restantes elemen-
tos de la misma forma (se supone que L1 es una lista de la forma [N1-E1,...,Nm-Em]).
Por ejemplo,

?- codificada_reducida_aux([1-a,2-b,3-a,1-c,3-b],L).
L = [a, 2-b, 3-a, ¢, 3-b]

La definiciéon de codificada_reducida_aux es

codificada_reducida_aux([1,[]).
codificada_reducida_aux([1-X|L1], [X|L2]) :-
codificada_reducida_aux(L1,L2).
codificada_reducida_aux([N-X|L1], [N-X|L2]) :-
N>1,
codificada_reducida_aux(L1,L2).

La definicién anterior puede simplificarse introduciendo un corte:

codificada_reducida_aux([],[]).

codificada_reducida_aux([1-X|L1],[X|L2]) :-
|

codificada_reducida_aux(L1,L2).
codificada_reducida_aux([N-X|L1], [N-X|L2]) :-

% N> 1,

codificada_reducida_aux(L1,L2).

4.23. Decodificacion de lista

Ejercicio 4.25 Definir la relacién decodificada(+L1,-L2) que, dada la lista L1, devuelve la
lista L2 cuya codificacion reducida por longitud es L1. Por ejemplo,

?- decodificada([a,2-b,3-a,c,3-b],L).
L=7[a, b, b, a, a, a, ¢, b, b, b]

Solucidon: La definicién de decodificada es

decodificada([]1,[]).

decodificada([1-X|L1], [X|L2]) :-
|

3

decodificada(L1,L2).
decodificada([N-X|L1],[X|L2]) :-

%N > 1,

"

N1 is N - 1,

decodificada([N1-X|L1],L2).
decodificada([X|L1], [XIL2]) :-

% X es atémico

decodificada(L1,L2).

4,24, Codificacion reducida directa

Ejercicio 4.26 Definir la relacién codificada_directa(+L1,-L2) que se verifica si L2 es la
codificacién reducida de L1 (es decir, las sucesiones de un mismo elemento X de L1 se codifican
por términos de la forma N-X donde N es la longitud de la sucesion cuando N es mayor que 1y
por X cuando N es igual a 1), pero en su definicion no se crean listas de elementos repetidos como
en la definicion de codificada_reducida. Por ejemplo,

?- codificada_directa([a,b,b,a,a,a,c,b,b,b],L).
L = [a, 2-b, 3-a, ¢, 3-b] ;

Solucidon: La definicién de codificada_directa es

codificada_directa([],[]).

codificada_directa([X|L1], [TIL2]) :-
cuenta_y_resto(X, [X|L1],N,L3),
término(N,X,T),
codificada_directa(L3,L2).

La relacién cuenta_y_resto(+X,+L1,-N,-L2) se verifica si N es el nimero de veces
que aparece X en la cabeza de la lista L1 y L2 es el resto de la lista L1 cuando se le quita
la sucesion de elementos X de su cabeza. Por ejemplo,

7- cuenta_y_resto(b, [b,b,a,a,a,c,b,b,b],N,L).
N =2
L =7[a, a, a, ¢, b, b, b]

La definiciéon de cuenta_y_resto es

cuenta_y_resto(X, [X|L1],N,L2) :-
cuenta_y_resto(X,L1,M,L2),

N is M+1.
cuenta_y_resto(X, [YIL],0, [YIL]) :-
X \=Y.

cuenta_y_resto(_,[],0,[1).

La definicién anterior puede simplificarse con cortes:

cuenta_y_resto_1(X, [X|L1],N,L2) :-
|

cuenta_y_resto_1(X,L1,M,L2),
N is M+1.
cuenta_y_resto_1(_,L,0,L).

La relaciéon término (+N,+X,-T) se verifica si T es el término correspondiente al na-
mero Ny al elemento X (es decir, Tes X si N es 1 y es N-X en otro caso). Por ejemplo,

?- término(1,a,T).

T=a
?7- término(2,a,T).
T = 2-a

La definicién de término es

término(1,X,X).
término(N,X,N-X) :-
N> 1.

La definicién anterior puede simplificarse con cortes:

término_1(1,X,X) :- !,
término_1(N,X,N-X). % :- N > 1.

4.25. Cota superior de una lista de nameros

Ejercicio 4.27 Definir la relacién cota_superior (+L,+N) que se verifique si N es una cota
superior de L (es decir, todos los elementos de L son menores o iguales que N). Por ejemplo,

7- cota_superior([1,5,3],7).
Yes

?7- cota_superior([1,5,3],5).
Yes

?7- cota_superior([1,5,3],4).
No

Dar dos definiciones, una recursiva y la otra no recursiva.

Solucién: La definicién recursiva de cota_superior es

cota_superior_1([],_).

cota_superior_1([X|L],N) :-
X =< N,
cota_superior_1(L,N).

La definicién no recursiva de cota_superior es

cota_superior_2(L,N) :-
\+ (member (X,L), X > N).

Mediante el siguiente ejemplo se compara la eficiencia de las dos definiciones

?7- numlist(1,10000,_L), time(cota_superior_1(_L,10000)).

% 20,001 inferences, 0.00 CPU in 0.01 seconds (0% CPU, Infinite Lips)
7- numlist(1,10000,_L), time(cota_superior_2(_L,10000)).

% 30,002 inferences, 0.01 CPU in 0.01 seconds (937 CPU, 3000200 Lips)
?7- numlist(1,100000,_L), time(cota_superior_1(_L,100000)).

% 200,001 inferences, 0.07 CPU in 0.07 seconds (99% CPU, 2857157 Lips)
7- numlist(1,100000,_L), time(cota_superior_2(_L,100000)).

% 300,002 inferences, 0.08 CPU in 0.08 seconds (1047 CPU, 3750025 Lips)

4.26. Dientes de sierra

Ejercicio 4.28 Definir la relacién diente(+L,-L1,-X,-L2) que se verifique si L se compone
de una lista L1 de niimeros estrictamente creciente hasta un cierto niimero X que llamaremos
cima, de la cima y de una lista L2 de niimeros estrictamente decreciente. Las listas tiene L1 y L2
tienen que ser no vacias y la cima X es el mayor elemento de L. Por ejemplo,

?- diente([1,2,5,4,3,1],L1,X,L2).
L1 = [1, 2]

X=5

L2 = [4, 3, 1] ;

No

?- diente([1,2,5],L1,X,L2).

No

Las listas que poseen esta forma de descomposicion se llaman dientes.

Solucidon: La definicidon de diente es

diente(L, [X1|L1],X, [X2|L2]) :-
append ([X1|L1], [X,X2(L2],L),
creciente([X1|L1]),
last([X1|L1],Y),
Y < X,
decreciente([X,X2|L2]).

La relacion creciente (+L) se verifica si la lista de ntimeros L es estrictamente creciente.

creciente([_]).

creciente([A,BIL]) :-
A < B,
creciente([BIL]).

La relacién decreciente (+L) se verifica si la lista de ntimeros L es estrictamente decre-
ciente.

decreciente([_]).

decreciente([A,B[L]) :-
A > B,
decreciente([B|L]).

Ejercicio 4.29 Una sierra es una lista numérica compuesta por la yuxtaposicion de dientes.
Notese que dos dientes consecutivos deben compartir un elemento. Por ejemplo [1,2,1,3,1] es
una sierra compuesta por los dientes [1,2,1] y [1,3,1], pero [1,2,1,1,3,1] no es una sierra.

Definir la relacion dientes_de_sierra(+L1,7L2) que se verifique si L1 es una sierra y
L2 es la lista de los dientes de L1. Por ejemplo,

?- dientes_de_sierra([1,2,1,3,1],L).

L =[[1, 2, 1], [1, 3, 111 ;

No

?- dientes_de_sierra([1,2,1,1,3,1],L).
No

Solucidon: La definicion de dientes_de_sierra es

dientes_de_sierra(L,[L]) :-
diente(L,_,_,_), !.

dientes_de_sierra(L, [L1|R]) :-
append(L1,L2,L),
diente(L1,_,_,_),
last(L1,X),
dientes_de_sierra([X|L2],R).

El corte aumenta la eficiencia como se aprecia en el siguiente ejemplo con la definicién
con corte

?7- numlist(1,100,_L1), append(_L1,[1|_L1],_L2), time(dientes_de_sierra(_L2,L)).
% 581,868 inferences, 0.36 CPU in 0.37 seconds (967 CPU, 1616300 Lips)
No

y el mismo ejemplo con la definicién sin el corte

?7- numlist(1,100,_L1), append(_L1,[1|_L1],_L2), time(dientes_de_sierra(_L2,L)).
% 3,147,270 inferences, 0.92 CPU in 0.94 seconds (98% CPU, 3420946 Lips)
No



Capitulo 5

Programacion logica de segundo orden

5.1. Determinaciéon de un niimero por su factorial

Ejercicio 5.1 Definir la relacion factorial_inverso (+X,-N) que se verifique si X es el facto-
rial de N. Por ejemplo,

?- factorial_inverso(120,N).

N =5,

No

?- factorial_inverso(80,N).
No

Solucién: Presentamos tres definiciones y comparamos su eficicencia.
La primera definicién usa un acumulador para almacenar los candidatos de la so-
lucién.

factorial_inverso_1(X,N) :-
factorial_inverso_1_aux(X,1,N).

La relacién factorial_inverso_1_aux(+X,+A,-N) se verifica si N es el menor nimero
mayor o igual que A cuyo factorial es X.

factorial_inverso_1_aux(X,A,A) :-
factorial(A,X).
factorial_inverso_1_aux(X,A,N) :-
factorial(A,N1),
N1 < X,
Al is A + 1,
factorial_inverso_1_aux(X,A1,N).

77

La relaciéon factorial (+N, -X) se verifica si X es el factorial de N.

factorial(l,1).
factorial(N,X) :-
N>1,
N1 is N-1,
factorial(N1,X1),
X is X1 * N.

La segunda definicién se diferencia de la anterior en que almacena en memoria los
factoriales de los candidatos considerados.

factorial_inverso_2(X,N) :-
factorial_inverso_2_aux(X,1,N).

La relacion factorial_inverso_2_aux(+X,+A,-N) se verifica si N es el menor niimero
mayor o igual que A cuyo factorial (con memoria) es X.

factorial_inverso_2_aux(X,A,A) :-
factorial_con_memoria(A,X).
factorial_inverso_2_aux(X,A,N) :-
factorial_con_memoria(A,N1),
N1 < X,
Cl is A + 1,
factorial_inverso_2_aux(X,C1,N).

La relacién factorial_con_memoria(+N, -X) se verifica si X es el factorial de N. Ademads
almacena en la base de datos internas los factoriales calculados.

:- dynamic factorial_con_memoria/2.

factorial_con_memoria(l,1).
factorial_con_memoria(N,X) :-
N>1,
N1 is N-1,
factorial_con_memoria(N1,X1),
X is X1 * N,
asserta(factorial_con_memoria(N,X) :- !').

En la tercera definicion se utiliza dos acumuladores para almacenar el candidato y
el factorial del candidato anterior.

factorial_inverso_3(X,N) :-
factorial_inverso_aux_3(X,1,1,N).

La relaciéon factorial_inverso_aux_3(+X,+A,+F,-N) se verificasi X = A% (A +1)

-+ * N (de forma que si A = 1 entonces X = N!).

factorial_inverso_aux_3(X,A,F,A) :-
A*F =:= X.
factorial_inverso_aux_3(X,A,F,N) :-
F1 is AxF,
F1 <X, !,
Al is A + 1,
factorial_inverso_aux_3(X,A1,F1,N).

En los siguientes ejemplos se compara la eficiencia.

?7- factorial(250,_X), time(factorial_inverso_1(_X,_N)).
% 249,501 inferences, 0.28 CPU in 0.30 seconds (94% CPU, 891075 Lips)

?7- factorial(250,_X), time(factorial_inverso_2(_X,_N)).
% 3,487 inferences, 0.00 CPU in 0.00 seconds (0% CPU, Infinite Lips)

?7- factorial(250,_X), time(factorial_inverso_3(_X,_N)).
% 1,248 inferences, 0.00 CPU in 0.01 seconds (0% CPU, Infinite Lips)

?- factorial(1000,_X), time(factorial_inverso_1(_X,_N)).
% 321,722 inferences, 0.46 CPU in 0.47 seconds (98% CPU, 699396 Lips)
ERROR: Out of local stack

?- factorial(1000,_X), time(factorial_inverso_2(_X,_N)).
% 13,987 inferences, 0.07 CPU in 0.07 seconds (967 CPU, 199814 Lips)

?7- factorial(1000,_X), time(factorial_inverso_3(_X,_N)).
% 4,998 inferences, 0.03 CPU in 0.03 seconds (1047 CPU, 166600 Lips)

5.2. Arbol de resolucién y definiciones equivalentes

Ejercicio 5.2 Se piden los siguientes apartados:

1. Dibujar el drbol de resolucion correspondiente al programa

p([],[]).

p([X/4],[X|B]) :-
X> 4,1,
p(4,B).

p([X|4],B) :-
p(4,B).

y al objetivo
| 7- p([5,1,61,B).
2. Explicar la relacion que hay entre L1 y L2 cuando se verifica p(L1,L2).

3. Dar una definicion no recursiva del predicado p(L1,L2).

Solucidén:
1. El &rbol de resolucién estd en la figura Bl (pagina BT)).

2. la relaciéon p(L1,L2) se verifica si L2 es la lista de los elementos de L1 que son
mayores que 4.

3. Una definicién no recursiva de p es

p1(L1,L2) :-
findall(X, (member (X,L1),X>4),L2).

5.3. Nodos de una generacién en una lista de arboles bi-
narios

Ejercicio 5.3 Un drbol binario es vacio o consta de tres partes: la raiz (que debe de ser un niimero
positivo), el subdrbol izquierdo (que debe ser un drbol binario) y el subdrbol derecho (que debe ser
un drbol binario). Usaremos la siguiente representacion

= nil representa el drbol vacio
= t(I,R,D) representa el drbol de la raiz R, subdrbol izquierdo I y subdrbol derecho D.

Por ejemplo, t(t(nil,2,nil),1,t(t(nil,4,nil),3,nil)) representa el drbol

p([5,1,6],B0)

2 [X1/5, 3
AL[1,6],
BO/[5/B1]}

5>4,1,
p([1.6],B1)

|

p((L6],B1)

2{X3/1, 3{X3/1,
A3/[6], / \ A3/[6],
BL/[1|B3]} B3/B1}

;(FGT’E!a’s) p(6],B1)

2 {X4l6,
Fallo A4/, \
B1/[6|B4]}

6>4,!,
p((1.B4)

!
p((1.B4)

p([1.B4)

£ 1{B4/[}

B =B0=[5/B1] = [5,6/B4] = [5,6]

Figura 5.1: Arbol de resolucién

Definir la relacion generacién(+N,+L1,-L2) que se verifique si L2 es la lista de nodos de la
generacion N de la lista de drboles L1. Por ejemplo,

?7- generacioén(0, [t(t(nil,2,nil),3,nil),t(nil,4,t(nil,5,nil))],L).
L = [3, 4]

?7- generacién(l, [t(t(nil,2,nil),3,nil),t(nil,4,t(nil,5,nil))],L).
L = [2, 5]

?7- generacion(2, [t(t(nil,2,nil),3,nil),t(nil,4,t(nil,5,nil))],L).
L =[]

Solucién: La definicién de generacién es

generacién(0,L,G): -
findall(R,member(t(_,R,_),L),G).
generaci6on(N,L,G): -
N >0,
elimina_raices(L,L1),
N1 is N-1,
generaci6én(N1,L1,G).

donde elimina_raices(+L1,-L2) se verifica si L2 es la lista de los &rboles obtenidos de
la lista de &rboles L1 eliminando sus raices. Por ejemplo,

?- elimina_raices([t(t(nil,2,nil),3,nil),t(nil,4,t(nil,5,nil))],L).
L = [t(nil, 2, nil), nil, nil, t(nil, 5, nil)]

La definicién de elimina_raices es

elimina_raices([1,[1).

elimina_raices([nil|L1],L2):-
elimina_raices(L1,L2).

elimina_raices([t(I,_,D)|L1],[I,DIL2]):-
elimina_raices(L1,L2).

5.4. Lista de elementos tnicos

Ejercicio 5.4 Definir la relacién anicos (+L1,-L2) que se verifique si L2 es la lista de los ele-
mentos que ocurren solamente una vez en la lista L1. Por ejemplo,

?- tnicos([2,5,3,2],L).

L = [5, 3]
?- 1tnicos([2,5,5,2],L).
L =[]

Solucidon: La definicion de Gnicos es

dnicos(L1,L2) :-
findall(X,es_tinico(X,L1),L2).

donde la relacién es_iinico (7X,+L) se verifica si X es un elemento que ocurre solamente
una vez en la lista L. Por ejemplo,

?- es_iunico(X,[2,5,3,2]).

X =5
X=3;
No

es_unico(X,L) :-
select(X,L,R),
not (memberchk(X,R)).

5.5. Elementos mas frecuentes de una lista

Ejercicio 5.5 Definir el predicado populares(L1,L2) que se verifique si L2 es la lista de los
elementos de L1 que aparecen el mayor niimero de veces. Por ejemplo,

7- populares([rosa, juan,david,manu,rosa,nuria,david],L).
L = [david,rosal

Solucidén: La definicién de populares es

populares(L1,L2) :-
setof (X,
((member(X,L1),
cuenta(X,L1,N1),

not ((member(Y,L1),
cuenta(Y,L1,N2),
N1 < N2)))),
L2).

La relacién cuenta(+X,+L,-N) se verifica si N es el niimero de veces que aparece el ele-
mento X en la lista L. Por ejemplo,

?7- cuenta(d, [r,j,d,m,r,n,d],N).
N =2

La definicion de cuenta es

cuenta(_, [],0).
cuenta(A, [BIL],N) :-
A ==B, !,
cuenta(A,L,M),
N is M+1.
cuenta(A,[_BIL],N) :-
% A \== _B,
cuenta(A,L,N).

5.6. Problema3dn +1

Ejercicio 5.6 Consideremos la funcion siguiente definida sobre los niimeros naturales:

3x+1, sixesimpar;
flx) = .
x/2, si x es par

se pide:

1. Definir la relacion sucesién(+X,?L) que se verifique si L es la lista de los elementos X,
f(X), f(f(X)), ..., f"(X) tal que f"(X) = 1. Por ejemplo,

?7- sucesién(3,L).
L = [3, 10, 5, 16, 8, 4, 2, 1]

L se llama la sucesion generada por N.

2. Definir la relacién longitudes (+X,?L) que se verifica si L la lista de pares Y-N donde Y
es un nimero de 1 a X y N es la longitud de sucesion generada por Y. Por ejemplo,

?7- longitudes(5,L).
L = [1-1, 2-2, 3-8, 4-3, 5-6]

3. Definir la relacién longitud_max (+X,?P) que se verifica si P es un par de la forma Y-N
donde Y es un niimero entre 1 y X tal que la longitud de la sucesion generada por Y es la
mds larga de las sucesiones generadas por 1,2,...,X y N es la longitud de dicha sucesion.
Por ejemplo,

?7- longitud_max(10,L).
L = 9-20

4. Definir menor_que_genera_mayor (+N,-M) que se verifique si M es el menor niimero na-
tural tal que la longitud de la sucesion generada por M es mayor que N. Por ejemplo,

7- menor_que_genera_mayor (100,N) .
N = 27

Solucidén: La relaciéon £ (+X, -Y) se verifica si Y= (X).

f(X,Y) :-
X mod 2 =:
Y is X/2.
£f(X,Y) :-
% X mod 2
Y is 3*X+1.

1]
o

1}
~
n
o

Solucién del apartado 1: La definicién de sucesidn es

sucesién(1,[1]) :- !'.
sucesién(X, [XIL]) :-
%X =/=1,
f(X,Y),
sucesién(Y,L).

Solucién del apartado 2: Presentamos dos definiciones de longitudes.
La primera definicién de longitudes es recursiva.

longitudes(X,L) :-
longitudes_aux(X,L1),
reverse(L1,L).

longitudes_aux (1, [1-N]) :-
!

b

sucesién(1,L),

length(L,N).
longitudes_aux (X, [X-N|L]) :-

%X > 1,

sucesién(X,L1),

length(L1,N),

Y is X-1,

longitudes_aux(Y,L).

La segunda definiciéon de longitudes, usando findall, es

longitudes_2(X,L) :-
findall(Y-N, (between(1,X,Y),sucesién(Y,S),length(S,N)),L).

Solucién del apartado 3: La definicién de longitud_max es

longitud_max(X,Y-N) :-
longitudes(X,L),
member (Y-N,L),
\+ (member(_Z-M,L), M > N).

Solucién del apartado 4: La definicién de menor_que_genera_mayor es

menor_que_genera_mayor (N,M) :-
menor _que_genera_mayor _aux(N,1,M).
menor_que_genera_mayor _aux(N,M,M) :-
sucesidén(M,L),
length(L,X),
X>N, !.
menor_que_genera_mayor _aux(N,X,M) :-
Y is X+1,
menor _que_genera_mayor _aux (N,Y,M).

5.7. Ntumeros perfectos

En los ejercicio de esta seccién estudiamos los ntiimeros perfectos (es decir, iguales
a la suma de sus divisores propios) y conceptos relacionados.

Ejercicio 5.7 Definir la relacion divisores_propios(+N,-L) que se verifique si L es la lista
ordenada de los divisores propios del niimero N. Por ejemplo,

7- divisores_propios(42,L).
L=1[1, 2, 3, 6, 7, 14, 21]

Solucién: La definicién de divisores_propios es

divisores_propios(N,L) :-
N1 is N -1,
findall(X, (between(1,N1,X), 0 =:= N mod X),L).

Ejercicio 5.8 Definir la relacién suma_divisores_propios(+N,-S) que se verifique si S es la
suma de los divisores propios del niimero N. Por ejemplo,

7- suma_divisores_propios(42,S).

S = 54
?7- suma_divisores_propios(1,S).
S=0

Solucién: La definiciéon de suma_divisores_propios es

suma_divisores_propios(N,S) :-
divisores_propios(N,L),
suma_lista(L,S).

La relaciéon suma_lista(+L,-N) se verifica si N es la suma de los elementos de la lista L.

suma_lista([],0).
suma_lista([X|L],S) :-
suma_lista(L,S1),
S is X + S1.

Ejercicio 5.9 Clasificamos los niimeros naturales en tres tipos:
m N es de tipo a si N es mayor que la suma de sus divisores propios
» Nesde tipo b si N es igual que la suma de sus divisores propios
» N es de tipo c si N es menor que la suma de sus divisores propios
Definir la relacion tipo (+N, -T) que se verifique si T es el tipo del niimero N. Por ejemplo,

?7- tipo(10,T).

T = a
7- tipo(28,T).
T=5>,

?7- tipo(12,T).
T =¢c

Solucién: La definicion de tipo es

tipo(N,T) :-
suma_divisores_propios(N,S),
tipo_aux(N,S,T).

tipo_aux(N,S,a) :- N > S, !.
tipo_aux(N,N,b) :- !.
tipo_aux(N,S,c). % :- N < S.

Ejercicio 5.10 Definir la relacion clasifica(+N,-L) que se verifique si L es la lista de tipos
de los miimeros comprendidos entre 1 y N. Por ejemplo,

?- clasifica(20,L).
L=7[a, a, a, a, a, b, a, a, a, a, a, ¢, a, a, a, a, a, ¢, a, cl

Solucion: La definicion de clasifica es

clasifica(N,L) :-
findall(T, (between(1,N,X),tipo(X,T)),L).

Ejercicio 5.11 Definir la relacién promedio (+N,-A,-B,-C) que se verifique si A, By C son las
cantidades de niimeros naturales menores o iguales que N de tipo a, b y c, respectivamente. Por
ejemplo,

?7- promedio(20,A,B,C).

A =16
B=1
Cc=3

Solucién: La definicién de promedio es

promedio(N,A,B,C) :-
clasifica(N,L),
promedio_aux(L,A,B,C).

promedio_aux([],0,0,0).
promedio_aux([a|L],A1,B,C) :-
promedio_aux(L,A,B,C),
Al is A+1.
promedio_aux([b|L],A,B1,C) :-
promedio_aux(L,A,B,C),

Bl is B+1.
promedio_aux([c|L],A,B,C1) :-
promedio_aux(L,A,B,C),

C1l is C+1.

Ejercicio 5.12 Definir la relacion menor (+N, -X) que se verifique si X es el menor niimero tal
que la cantidad de niimeros naturales menores o iguales que X de tipo a es N. Por ejemplo,

?- menor(20,X).
X =25

Soluciéon: La definicién de menor es

menor (N,X) :-
menor_aux(N,N,X) .

menor_aux(N,M,M) :-
promedio(M,N,_,_), !.
menor_aux(N,M,X) :-
M1 is M+1,
menor_aux(N,M1,X).

5.8. Determinacion de tridngulos equilateros

Ejercicio 5.13 Un poligono se representa por su nombre y las longitudes de sus lados. Definir
la relacion es_equilatero (+P) que se verifica si el poligono P es equildtero (es decir, que todos
sus lados son iguales). Por ejemplo,

7- es_equilatero(triangulo(4,4,4)).

Yes

7- es_equilédtero(cuadrilatero(3,4,5,3)).
No

Solucidén: La definicién de es_equilatero es

es_equilatero(P) :-
P =.. [_IL],
todos_iguales(L).

La relacion todos_iguales esta definida en la pagina

5.9. Operacion binaria aplicada a listas

Ejercicio 5.14 Definir la relacion operacién_listas(+0,+L1,+L2,-L3) que se verifica si L3
es la lista obtenida aplicando la operacion binaria O a los elementos de L1 y L2 que ocupan la
misma posicion. Por ejemplo,

?7- operacién_lista(+,[1,2,3],[4,5,6],L).
L =1[5,7, 9]

?7- operacidn_lista(x*,[1,2,3],[4,5,6],L).
L = [4, 10, 18]

Nota: Se supone que L1 y L2 tienen la misma longitud)

Solucién: La definicién de operacidn_lista es

operacidon_lista(_,[]1,[],[]1).
operacién_lista(0,[X1|L1], [X2]L2],[X3|L3]) :-
E =.. [0,X1,X2],
X3 is E,
operaci6én_lista(0,L1,L2,L3).

5.10. Numeros en un término

Ejercicio 5.15 Definir la relacion nimeros (+T,-L) que que se verifique si L es el conjunto de
todos los niimeros que ocurren en el término cerrado T. Por ejemplo,

?- numeros(f(a,g(c,5),2),L).

L = [5, 2]

?- ntmeros (a+3+b*(sen(2)+3),L).
L = [2, 3]

?- numeros(a+b,L).

L =[]

Solucidon: La definicidn de ntmeros es

nimeros (T, [T]) :-
number (T), !.
nimeros(T,L) :-
% not (number(T)),
T =.. [_IL1],
nimeros_de_lista(L1,L).

La relacién nimeros_de_lista(+L1,-L2) se verifica si L2 es el conjunto de ntimeros en
la lista de términos L1. Por ejemplo,

?7- numeros_de_lista([a+3, b*(sen(2)+3)],L).
L = [2, 3]

La definicién de ntmeros_de_lista es

nimeros_de_lista([],[]).

nimeros_de_lista([T|L1],L2) :-
ntumeros (T,L3),
nimeros_de_lista(L1,L4),
union(L3,L4,L2).

5.11. Palabra sin vocales

Ejercicio 5.16 Definir la relacién elimina_vocales(+P1,-P2) que se verifique si P2 es la
palabra que se obtiene al eliminar todas las vocales de la palabra P1. Por ejemplo,

?7- elimina_vocales(sevillano,P).
P = svlln

Solucidon: La definicion de elimina_vocales es

elimina_vocales(P1,P2) :-
name (P1,L1),
coédigos_vocales(L),
findall (N, (member (N,L1) ,not(member(N,L))),L2),
name (P2,L2) .

La relacién cédigos_vocales(?L) se verifica si L es la lista de los c6digos ASCII de las
vocales.

cédigos_vocales(L) :-
name (aeiou,L).

5.12. Palabras maximales

Ejercicio 5.17 Definir la relacion longitud (+P,-N) que se verifique si N es la longitud de la
palabra P. Por ejemplo,

7- longitud(ana,N).
N =23

Solucién: La definicién de longitud es

longitud(P,N) :-
name (P,L),
length(L,N).

Ejercicio 5.18 Definir la relacién palabra_maximal (+L,-P) que se verifique si P es una pala-
bra maximal (es decir, de mdxima longitud) de la lista de palabras L. Por ejemplo,

7- palabra_maximal([eva,y,ana,se,van],P).
P = eva ;

P = ana ;

P = van ;

No

Solucién: La definicién de palabra_maximal es

palabra_maximal(L,P) :-
select(P,L,R),
longitud(P,N),
not ((member(P1,R), longitud(P1,N1), N < N1)).

Ejercicio 5.19 Definir la relacién palabras_maximales(+L1,-L2) que se verifique si L2 es la
lista de las palabras maximales de la lista de palabras L1. Por ejemplo,

7- palabras_maximales([eva,y,ana,se,van],L).
L = [eva, ana, van]

Solucidén: La definicién de palabras_maximales es

palabras_maximales(L1,L2) :-
findall(P,palabra_maximal(L1,P),L2).

5.13. Clausura transitiva de una relacion

Ejercicio 5.20 La clausura transitiva de una relacion binaria R es la menor relacion transitiva
que contiene a R; por ejemplo, la clausura transitiva de {(a,b), (b,c)} es {(a,b), (b,c), (a,c)}.
Definir el predicado clausura_transitiva(R,X,Y) que se verifique si (X,Y) estd en la clau-
sura transitiva de la relacion R. Por ejemplo, suponiendo que se han definido las relaciones p y q

por

p(a,b).
p(b,c).
g(c,b).
q(b,a).

se tiene

?7- clausura_transitiva(p,X,Y).
X =a Y=Db;

X=0D» Y=¢c¢ ;

X =a Y=¢c¢ ;

No

?7- clausura_transitiva(q,X,Y).
X=c¢c Y=Db;

X=0D» Y =a ;

X=c¢ Y =a ;

No

Solucidon: La definicidon de clausura_transitiva es

clausura_transitiva(R,X,Y) :-
apply (R, [X,Y]).

clausura_transitiva(R,X,Y) :-
apply (R, [X,Z]),
clausura_transitiva(R,Z,Y).

5.14. Traduccién de cifras a palabras

Ejercicio 5.21 Definir la relacién traduccidén(+L1,-L2) que se verifique si L2 es la lista de
palabras correspondientes a los digitos de la lista L1. Por ejemplo,

?- traduccidén([1,3],L).
L = [uno, tres]

(Indicacion: Usar la relacion auxiliar nombre (D,N) que se verifica si N es el nombre del digito
D).

Soluciéon: La definicion de la relacion auxiliar nombre es

nombre (0, cero) .
nombre (1,uno) .
nombre(2,dos) .
nombre(3,tres).
nombre (4, cuatro) .
nombre (5,cinco) .
nombre(6,seis).
nombre(7,siete).
nombre (8, ocho) .
nombre (9,nueve) .

Se presentan tres definiciones de traduccién.
Primera solucién: Una definicién de traduccién(L1,L2), por recursién en L1, es

traduccién_1([1,[1).

traduccién_1([D|L1],[N|L2]) :-
nombre(D,N),
traduccién_1(L1,L2).

Segunda solucién: Una definicién de traduccién usando findall es

traduccién_2(L1,L2) :-
findall(N, (member(D,L1) ,nombre(D,N)),L2).

Tercera soluciéon: Una definicion de traduccién usando maplist es

traduccién_3(L1,L2) :-
maplist (nombre,L1,L2).

5.15. Transformacidn de lista dependiente de la posicién

Ejercicio 5.22 Definir la relacion transforma(+L1,-L2) que se verifique si L2 es la lista obte-
nida sumdndole a cada elemento numérico de L1 su posicion en la lista. Por ejemplo,

?- transforma([1,1,1,a,b,c,1,1,1],L).
L=1[2, 3, 4, a, b, c, 8, 9, 10]

?- transforma([1,2,a,5,2,b,3,1],L).
L=1[2, 4, a, 9, 7, b, 10, 9]

Dar dos definiciones, una recursiva y otra no—recursiva.

Solucion: La definiciéon de recursiva de transforma es

transforma(L1,L2) :-
transforma_aux(L1,1,L2).

donde transforma_aux(+L1,+N,-L2) se verifica si L2 es la lista obtenida afiadiéndole a
cada elemento numérico de L1 la suma de N y su posicién en la lista. Por ejemplo,

transforma_aux([],_,[]).
transforma_aux([X|L1],N,[YIL2]) :-
number (X), !,
Y is X+N,
N1 is N+1,

transforma_aux(L1,N1,L2).
transforma_aux([X|L1],N, [X|L2]) :-

% not (number (X)),

N1 is N+1,

transforma_aux(L1,N1,L2).

La definicién no recursiva de transforma es

transforma_2(L1,L2) :-
lista_de_posiciones(L1,L),
maplist (suma,L1,L,L2).

donde lista_de_posiciones(+L1,-L2) se verifica si L2 es la lista de posiciones corres-
pondiente a la lista L1. Por ejemplo,

7- lista_de_posiciones([1,1,1,a,b,c,1,1,1],L).
L=1[1, 2, 3, 4, 5, 6, 7, 8, 9]

lista_de_posiciones(L1,L2) :-
length(L1,N),
findall(X,between(1,N,X),L2).

y suma(+X,+Y,-Z) se verifica si Z es la suma de X y el nimero Y, cuando X es un ntimero
y es igual a X, en caso contrario. Por ejemplo,

7- suma(3,2,Z).
Z =5
?- suma(b,2,Z).
Z ="

suma(X,Y,Z) :-
number (X), !,
Z is X+Y.

suma (X, _,X).

5.16. Aplanamiento de listas

Ejercicio 5.23 Definir la relacién aplana(+L1,?L2) que se verifique si L2 es la lista obtenida
reemplazando, recursivamente, cada lista de L1 por sus elementos. Por ejemplo,

?- aplana([a, [b, [c]],[[d],el],L).
L =1[a, b, c, d, el

Solucién: Para definir aplana(X,Y) vamos a generalizar su dominio de forma que
si X no es una lista, entonces Y es la lista cuyo tinico elemento es X.

aplana(X, [X]) :-

\+ is_list(X).
aplana([],[1).
aplana([X|L1],L2) :-

aplana(X,L3),

aplana(L1,L4),
append(L3,L4,L2).




Capitulo 6

Estilo y eficiencia en programacion
logica

6.1. Numero de Hardy

En cierta ocasién, el mateméatico Ramanujan estaba en un hospital en Inglaterra y su
amigo Hardy fue a visitarlo. Hardy coment6 que habia llegado al hospital en un taxi de
matricula N y esperaba que éste no fuese un mal presagio, ya que N era un ndmero poco
interesante. Ramanujan no estuvo de acuerdo ya que inmediatamente dijo que N tiene
una propiedad muy especial: N es el menor entero positivo que puede descomponerse
de dos maneras distintas como suma de dos cubos.

El objetivo de esta seccién es averiguar la matricula del taxi que llevé a Hardy a
visitar a Ramanujan.

Ejercicio 6.1 Definir la relacién es_cubo (+N) que se verifique si N es el cubo de un entero. Por
ejemplo,

7- es_cubo_1(1000).
Yes
?7- es_cubo_1(1001).
No

Solucién: Presentaremos distintas definiciones y comentaremos su eficiencia.
La primera solucién realiza una btisqueda desde 1 hasta N.

es_cubo_1(N) :-
between(1,N,X),
N is X*Xx*X.

La segunda solucion realiza una bsqueda desde 1 hasta v/N.

97

es_cubo_2(N) :-
Cota is round(N~(1/3)),
between(1,Cota,X),
N is X*XxX.

La tercera solucion utiliza predicados aritméticos predefinidos.

es_cubo_3(N) :-
N =:= round(N *x (1/3)) =*x 3.

Para comparar la eficiencia realizamos los siguientes experimentos:

7- time(es_cubo_1(1000001)).

% 1,000,003 inferences, 1.23 CPU in 1.25 seconds (99% CPU, 813011 Lips)
No

7- time(es_cubo_2(1000001)).

% 103 inferences, 0.00 CPU in 0.00 seconds (0% CPU, Infinite Lips)

No

?7- time(es_cubo_3(1000001)).

% 3 inferences, 0.00 CPU in 0.00 seconds (0% CPU, Infinite Lips)

No

Se observa que la més eficiente es la tercera definicién. La segunda se aproxima a la
segunda. La primera es muy ineficiente. En lo que sigue adoptaremos la tercera como
definicén de es_cubo.

es_cubo(N) :-
es_cubo_3(N).

Ejercicio 6.2 Definir la relacion descompone (+N, -X, -Y) que se verifique si X e Y son dos cubos
cuya suma es N y, ademds, X es menor o igual que Y. Por ejemplo,

7- descompone(1008,X,Y).

X =28
Y = 1000 ;
No

Solucién: Presentaremos distintas definiciones y comentaremos su eficiencia.
La primera definicion realiza una bisquedaen X e Y.

descompone_1(N,X,Y) :-
between(1,N,X),
between(1,N,Y),
es_cubo(X),
es_cubo(Y),
X =< Y,
N is X+Y.

La segunda definicion realiza la biisqueda en X y determina Y.

descompone_2(N,X,Y) :-
between(1,N,X),
es_cubo(X),

Y is N - X,
X =< Y,
es_cubo(Y).

La tercera definicion realiza una busqueda acotada.

descompone_3(N,X,Y) :-
Cota is round((N/2)~(1/3)),
between(1,Cota,M),
X is MxM%*M,
Y is N-X,
X =<Y,
es_cubo(Y).

Para comparar la eficiencia realizamos los siguientes experimentos:

7- time(descompone_1(1707,X,Y)).

% 11,732,455 inferences, 7.89 CPU in 7.91 seconds (100% CPU, 1487003 Lips)
No

7- time(descompone_2(1707,X,Y)).

% 6,890 inferences, 0.01 CPU in 0.01 seconds (112% CPU, 689000 Lips)

No

7- time(descompone_3(1707,X,Y)).

% 66 inferences, 0.00 CPU in 0.00 seconds (0% CPU, Infinite Lips)

No

7- time(descompone_1(2331,X,Y)).

% 9,410,498 inferences, 6.31 CPU in 6.33 seconds (1007 CPU, 1491363 Lips)

X = 1000
Y = 1331
Yes

7- time(descompone_2(2331,X,Y)).
% 4,062 inferences, 0.01 CPU in 0.01 seconds (189% CPU, 406200 Lips)

X = 1000
Y = 1331
Yes

7- time(descompone_3(2331,X,Y)).
% 73 inferences, 0.00 CPU in 0.00 seconds (0% CPU, Infinite Lips)

X = 1000
Y = 1331
Yes

Se observa que la mads eficiente es la tercera definicién. En lo que sigue adoptaremos la
tercera como definicién de descompone.

descompone (N,X,Y) :-
descompone_3(N,X,Y).

Ejercicio 6.3 Definir la relacion ramanujan(+N) que se verifique si N puede descomponerse en
suma de dos cubos exactamente de dos maneras distintas.

Solucién: La definiciéon de ramanujan es

ramanujan(N) :-
setof (par(X,Y),descompone(N,X,Y),[_,_1).

Ejercicio 6.4 Definir la relacion hardy (-N) que se verifique si N es el menor entero positivo que
satisface el predicado ramanujan anterior. ;Cudl es la la matricula del taxi que llevé a Hardy a
visitar a Ramanujan?

Solucién: La definiciéon de hardy es

hardy(N) :-
hardy_aux(N,1).

hardy_aux(N,N) :-
ramanujan(N),
!

hardy_aux(N,M) :-
M1 is M+1,
hardy_aux(N,M1).

La matricula del taxi que llev6 a Hardy a visitar a Ramanujan se calcula mediante
la siguiente consulta

?- hardy(N) .
N = 1729

Por tanto, la matricula del taxi es 1729.

6.2. Subconjuntos de suma dada

Ejercicio 6.5 Definir la relacién subconjunto_suma(+L1,+N,?L2) que se verifique si L2 es
un subconjunto de L1 tal que la suma de los elementos de L2 es N. Por ejemplo,

?7- subconjunto_suma([10,7,3,4,2,1],7,L).
L = [7] ;

L = [3, 4] ;

L= [4, 2, 1] ;

No

?- subconjunto_suma([1,2,3],0,L).
L=1[;

No

Solucién: Presentaremos cuatro definiciones y comparemos su eficiencia.
La primera definicion usa la relacién subconjunto definida en el ejercicio [LTZ (pé-

gina [20).

subconjunto_suma_1(L1,N,L2) :-
subconjunto(L1,L2),
sumlist (L2,N).

En la segunda definicion se adapta la definicién de subconjunto teniendo en cuenta
la suma de sus elementos.

subconjunto_suma_2([1,0,[]1).
subconjunto_suma_2([X|L1],N, [X|L2]) :-
N >= X,
N1 is N-X,
subconjunto_suma_2(L1,N1,L2).
subconjunto_suma_2([_|L1],N,L2) :-
subconjunto_suma_2(L1,N,L2).




102 Capitulo 6. Estilo y eficiencia en programacién légica

En la tercera definicion se define de forma dindmica la relacién subconjuntos (+L1,+N.-S)

que se verifica si S es la lista de subconjuntos de S tales que la suma de sus elementos es
N.

:- dynamic subconjuntos_suma_calculados_3/3.

subconjunto_suma_3(L1,N,L2) :-
subconjuntos_suma_calculados_3(L1,N,S), !
member (L2,S) .

subconjunto_suma_3(L1,N,L2) :-
findall(L,subconjunto_suma_3_aux(L1,N,L),S),
asserta(subconjuntos_suma_calculados_3(L1,N,S)),
member (L2,S) .

3

subconjunto_suma_3_aux([]1,0,[]).
subconjunto_suma_3_aux([X|L1],N, [X|L2]) :-
N >= X,
N1 is N-X,
subconjunto_suma_3(L1,N1,L2).
subconjunto_suma_3_aux([_|L1],N,L2) :-
subconjunto_suma_3(L1,N,L2).

La cuarta definicién es una variacién de la anterior de forma que se intercambian los

dos primeros argumentos de subconjuntos_suma_calculados a fin de que el argumento
sobre el que se indexa sea N.

:- dynamic subconjuntos_suma_calculados_4/3.

subconjunto_suma_4(L1,N,L2) :-
subconjuntos_suma_calculados_4(N,L1,S), !
member (L2,S) .

subconjunto_suma_4(L1,N,L2) :-
findall(L,subconjunto_suma_4_aux(L1,N,L),S),
asserta(subconjuntos_suma_calculados_4(N,L1,S)),
member (L2,8S) .

b

subconjunto_suma_4_aux([],0,[]).
subconjunto_suma_4_aux([X|L1],N, [X|L2]) :-
N >= X,
N1 is N-X,
subconjunto_suma_4(L1,N1,L2).

subconjunto_suma_4_aux([_|L1],N,L2) :-
subconjunto_suma_4(L1,N,L2).

Para comparar la eficiencia se realizan los siguientes experimentos.

7- numlist(1,20,_L1), time(findall(L,subconjunto_suma_1(_L1,10,L),_S)).
% 25,165,844 inferences, 9.09 CPU in 9.19 seconds

?7- numlist(1,20,_L1), time(findall(L,subconjunto_suma_2(_L1,10,L),_S)).
% 1,974 inferences, 0.00 CPU in 0.00 seconds

7- numlist(1,20,_L1), time(findall(L,subconjunto_suma_3(_L1,10,L),_S)).
% 3,242 inferences, 0.01 CPU in 0.01 seconds

7- numlist(1,20,_L1), time(findall(L,subconjunto_suma_4(_L1,10,L),_S)).
% 3,242 inferences, 0.01 CPU in 0.01 seconds
?7- numlist(1,140,_L1), time(findall(L,subconjunto_suma_2(_L1,70,L),_S)).

% 104,855,949 inferences, 47.90 CPU in 58.77 seconds

7- numlist(1,140,_L1), time(findall(L,subconjunto_suma_3(_L1,70,L),_S)).
% 593,122 inferences, 18.91 CPU in 21.54 seconds

7- numlist(1,140,_L1), time(findall(L,subconjunto_suma_4(_L1,70,L),_S)).
% 593,158 inferences, 1.96 CPU in 2.18 seconds

Se observa que la més eficiente es la cuarta definicion.

6.3. Coloreado de mapas

Ejercicio 6.6 Un mapa puede representarse mediante la relacion mapa (N, L) donde N es el nom-
bre del mapa y L es la lista de los pares formados por cada una de las regiones del mapa y la lista
de sus regiones vecinas. Por ejemplo, los mapas de la figura el se pueden representar por

mapa(ejemplo_1,
(a-[b,c,d], b-[a,d,e]l, c-[a,d,f], d-[a,b,c,e,f,g],
e-[b,d,gl, f-[c,d,gl, g-[d,e,f]11).

mapa(ejemplo_2,
[a-[b,e,k], b-[a,c,e,k], c-[b,d,f,k], d-[c,f,k], e-[a,b,g,k],
f-[c,d,h,k], g-[e,i,k], h-[f,j,k], i-[g,j,k1, j-[i,h,k],
k-[a,b,c,d,e,f,g,h,i,j1]1).

a b alb|ci|d

f g i J

Figura 6.1: Ejemplos de mapas

Definir la relacion coloracidén(+M,+LC, -S) que se verifique si S es una lista de pares formados
por una region del mapa My uno de los colores de la lista de colores LC tal que las regiones vecinas
tengan colores distintos. Por ejemplo,

?7- coloracién(ejemplo_1,[1,2,3],S).
S = [a-1, b-2, c-2, d-3, e-1, f-1, g—2]

¢ Qué niimero de colores se necesitan para colorear el segundo mapa? ;De cudntas formas dis-
tintas puede colorearse con dicho niimero?

Solucién: Presentamos dos definiciones y comparamos su eficiencia.
La primera definicién de coloracién es por generacién y prueba.

coloracién_1(M,LC,S) :-
mapa(M,L),
coloracién_1_aux(L,LC,S).

coloracidn_1_aux([1,_,[]1).
coloracién_1_aux([R-VI|L],LC, [R-C|S]) :-
member (C,LC),
coloracién_1_aux(L,LC,S),
not ((member (R1,V), member(R1-C,S))).

En la segunda definicién de coloracién se usa un acumulador y se adelanta las
pruebas.

coloracién_2(M,LC,S) :-
mapa(M,L),
coloracidén_2_aux(L,LC,[],S).

coloracién_2_aux([],_,S,S).

coloracién_2_aux([R-V|L],LC,A,S) :-
member (C,LC),
not ((member (R1,V), member(R1-C,A))),
coloracidén_2_aux(L,LC,[R-C|A],S).

Para comparar las dos definiciones usaremos el segundo mapa.

7- time(coloracidn_1(ejemplo_2,[1,2,3,4],S)).
% 16,705,318 inferences, 6.18 CPU in 6.20 seconds (100% CPU, 2703126 Lips)
S = [a-1, b-2, c-1, d-2, e-3, £-3, g-1, h-1, i-2, j-3, k-4]

?7- time(coloracidén_2(ejemplo_2,[1,2,3,4],S)).
% 546 inferences, 0.00 CPU in 0.00 seconds (0% CPU, Infinite Lips)
S = [k-4, j-3, i-2, h-1, g-1, £-3, e-3, d-2, c-1, b-2, a-1]

Se observa que la segunda es mads eficiente.
En el ejemplo anterior se observa que se puede colorear el segundo mapa con 4
colores. Veamos si puede colorearse con 3 colores.

‘ ?- coloracién_2(ejemplo_2,[1,2,3],S).
No

Por tanto, para colorear el segundo mapa se necesitan 4 colores.
El namero N de formas distintas de colorear el segundo mapa se calcula mediante
la siguiente consulta.

‘ ?7- findall(_S,coloracién_2(ejemplo_2,[1,2,3,4],_S),_L), length(_L,N).
N = 1032

Capitulo 7

Aplicaciones de programacion
declarativa

7.1. Formacidén de grupos minimales de asignaturas com-
patibles

Mediante la relaciéon lista_de_clase(C,A,L) se representa la informacién de los
alumnos segtn los cursos y asignaturas, de forma que C es el curso, A es la asignatura
y L es la lista de los alumnos de dicha asignatura. A lo largo del ejercicio vamos a usar
como ejemplo la siguiente informacién.

lista_de_clase(cl,asigl, [al,a5]).
lista_de_clase(cl,asig2,[al,a2,a3]).
lista_de_clase(cl,asig3,[al,a3]).
lista_de_clase(cl,asig4,[a2,a4,a6,a8]).
lista_de_clase(cl,asigh, [a2,a4,a5]).
lista_de_clase(cl,asig6, [a6,a7]).
lista_de_clase(cl,asig7, [a3,a7]).
lista_de_clase(cl,asig8,[a6,a7,a8]).
lista_de_clase(c2,asig9, [a9,all]).
lista_de_clase(c2,asigl10,[a10,a12]).
lista_de_clase(c2,asigll, [a10,a11]).
lista_de_clase(c2,asigl2, [a9,a12]).

Ejercicio 7.1 Definir la relacion asignaturas (+C,-L) que se verifique si L es la lista de asig-
naturas del curso C. Por ejemplo,

?7- asignaturas(ci1,L).
L = [asigl, asig2, asig3, asig4, asigb, asig6, asig7, asig8]

107

?7- asignaturas(c2,L).
L = [asig9, asiglO, asigll, asigl?2]

Solucién: La definicién de asignaturas es

asignaturas(C,L) :-
findall(X,lista_de_clase(C,X,_),L).

Ejercicio 7.2 Definir la relacion grupo_incompatible (+L) que se verifique si la lista de asig-
naturas L es incompatible (es decir, algiin alumno estd en las listas de clase de mds de una
asignatura de la lista L). Por ejemplo,

?- grupo_incompatible([asigl,asig2]).

Yes

7- grupo_incompatible([asigl,asig4,asig7]).
No

Solucién: La definicién de grupo_incompatible es

grupo_incompatible(L) :-
select(X,L,R),
member (Y,R),
lista_de_clase(_,X,LX),
lista_de_clase(_,Y,LY),
member (A,LX),
member (A,LY) .

Ejercicio 7.3 Definir la relacion 1lista_incompatible (+L) que verifique si la lista de grupos
de asignaturas L es incompatible (es decir, contiene algiin grupo incompatible de asignaturas).
Por ejemplo,

?7- lista_incompatible([[asig9, asigl2], [asigll, asigl0]]).
Yes
7- lista_incompatible([[asigll, asigl2], [asig9, asigl0]]).
No

Solucidén: La definicién de 1lista_incompatible es

lista_incompatible(P) :-
member (G,P),
grupo_incompatible(G) .

Ejercicio 7.4 Definir la relacién extensidn(+L1,+X,-L2) que se verifique si L2 es la lista obte-
nida afiadiendo X como primer elemento de un elemento de L1 o L2 es la lista obtenida afiadiendo
[X] a L1. Por ejemplo,

?7- extensién([[a], [b, c]1,d,E).

E = [[d, al, [b, cl] ;
E = [[al, [d, b, c]] ;
E = [[al, [b, c], [d]] ;
No

Solucidén: La definicidén de extensién es

extensién([],X, [[X]]).

extension([L|L1],X, [[XIL]IL1]).

extensidén([L|L1],X,[L|L2]) :-
extensién(L1,X,L2).

Ejercicio 7.5 Definir la relacién particién(+L,-P) que se verifique si P es una particion de la
lista L (es decir, un conjunto obtenido distribuyendo los elementos de L en conjuntos no vacios y
sin elementos comunes). Por ejemplo,

?7- particién([a,b,c],P).
P = [[a, b, cl] ;

P = [[b, c], [al] ;

P = [[a, c], [bl] ;

P = [[c], [a, bl] ;

P = [[c], [b]l, [al] ;
No

Solucién: La definicién de particidn es

particion([1,[]1).

particion([X|L1],L2) :-
particion(L1,L3),
extensidén(L3,X,L2).

Ejercicio 7.6 Definir la relacion agrupacién_compatible_de_asignaturas(+C,-P) que se
verifique si P es una particion compatible de las asignaturas del curso C. Por ejemplo,

7- agrupacién_compatible_de_asignaturas(c2,P).
P = [[asigll, asigl2], [asig9, asigl0l] ;

P = [[asigll, asigl2], [asigl0], [asig9]] ;
Yes



110 Capitulo 7. Aplicaciones de programacion declarativa

Solucién: La definicién de agrupacidén_compatible_de_asignaturas es

agrupacién_compatible_de_asignaturas(C,P) :-
asignaturas(C,L),
particién(L,P),
not(lista_incompatible(P)).

Ejercicio 7.7 Definir la relacion agrupacion_compatible_minimal (+C,-P) que se verifique
si P es una particion compatible de las asignaturas del curso C con el menor niimero posible de
grupos de asignaturas. Por ejemplo,

7- agrupacién_compatible_minimal(c2,P).
P = [[asigll, asigl2], [asig9, asiglOl] ;
No

Calcular las agrupaciones compatibles minimales del curso c1.

Solucién: La definicién agrupacidén_compatible_minimal es

agrupacién_compatible_minimal(C,P) :-
agrupacién_compatible_de_asignaturas(C,P),
length(P,N),
not ((agrupacidén_compatible_de_asignaturas(C,P1),length(P1,M) ,M < N)).

Las agrupaciones compatibles minimales del curso c1 se calculan mediante la con-
sulta

7- agrupacién_compatible_minimal(c1,P).

P = [[asig3, asigb, asig8], [asigl, asig4, asig7], [asig2, asig6l] ;
P = [[asig2, asig8], [asigl, asig4, asig7], [asig3, asigb, asig6l] ;
No

7.2. Simulacion de una calculadora basica

El objetivo de los siguientes ejercicios es la simulacién de una calculadora bésica.
Para ello consideraremos que en cada momento la calculadora se encuentra en un de-

terminado estado caracterizado por una lista con cuatro elementos [UCE,UTA,UOA, VIM]
donde

= UCE es el dltimo célculo efectuado,

= UTA es la ultima tecla activada,

= U0A es el tltimo operador activado y

= VIM es el valor impreso.

El estado inicial es [0,=,=,0] y estd definido por

estado_inicial([0,=,=,0]).

Las acciones posibles son pulsar un digito, una operacién aritmética o la de resul-
tado y estan definidas por

accion(X) :- es_digito(X).
accidén(X) :- es_operacidn(X).
accién(X) :- es_resultado(X).

es_digito(0).
es_digito(1).
es_digito(2).
es_digito(3).
es_digito(4).
es_digito(5).
es_digito(6).
es_digito(7).
es_digito(8).
es_digito(9).

es_operacidn(+).
es_operacidn(-).
es_operacidn(*).
es_operacidén(/).

es_resultado(=).

En la siguiente tabla se muestran los estados de la calculadora correspondientes a
las acciones indicadas en la taltima columna

estado tecla
(0, ==, 0|3
(0,3, =, 3)|+
(3, +, +, 3)|2
(3,2,+, 2)]|1
(3,1, +, 21) | *
(24, %, %, 24) | 2
(24, 2, %, 2)|=
(48, =, =, 48)

Es decir, si se parte del estado inicial y se realizan las acciones

3+21 %2

se obtiene como resultado el niimero 48.

Ejercicio 7.8 Definir la relacion transicion(+E1,+X,7E2) que se verifique si E2 es el estado
obtenido aplicando la accion X al estado E1; es decir, si E1 es [UCE,UTA,UOA,VIM], entonces

» Si X es un digito, entonces

e s5i UTA es un digito, E2 es [UCE,X,UOA, 10xVIM+X];
e ¢n otro caso, E2 es [UCE,X,U0A,X].

» Si X no es un digito, entonces

e si UOA es una operacion, E2 es [UOA(UCE,VIM),X,X,UOA(UCE,VIM)]
e ¢n otro caso, E2 es [VIM,X,X,VIM].

Por ejemplo,

El =
E2 =
E3 =
E4 =
Eb =

(o,
(o,
(3,
(3,
(3,

= N+ W

-

3

3

b

-

=, 0]
=, 3]
, 3]
, 2]
, 21]

+ + 4+

?7- estado_inicial(E1l),
transicién(E1,3,E2),
transicién(E2,+,E3),
transicién(E3,2,E4),
transicién(E4,1,E5),
transicidon(E5, *,E6),
transicion(E6,2,E7),
transicién(E7,=,E8).

E6 = [24, *, *, 24]
E7 = [24, 2, *, 2]
E8 = [48, =, =, 48]

Solucidén: Presentamos tres definiciones.
La primera definicién de transicién es

transicién_1([UCE,UTA,UOA,VIM],X,E) :-
( es_digito(X) ->
( es_digito(UTA) ->
Y is 10*%VIM+X,
E = [UCE,X,UQOA,Y]
; %h \+ es_digito(UTA) ->
E = [UCE,X,U0A,X] )
; h \+ es_digito(X) ->
( es_operacidén(UOA) ->
T =.. [UOA,UCE,VIM],
Y is T,
E = [Y,X,X,Y]
; % \+ es_operacién(U0A) ->
E = [VIM,X,X,VIM] )).

La segunda definiciéon de transicién es

transicién_2([UCE,UTA,UOA,VIM],X,E) :-

( es_digito(X), es_digito(UTA) ->
Y is 10*VIM+X,
E = [UCE,X,U0A,Y]

; es_digito(X), \+ es_digito(UTA) ->
E = [UCE,X,U0A,X]

; \+ es_digito(X), es_operaci6n(UOA) ->
T =.. [UOA,UCE,VIM],
Y is T,
E = [Y,X,X,Y]

; \+ es_digito(X), es_resultado(UOA) ->
E = [VIM,X,X,VIM] ).

La tercera definicién de transicién es

transicién_3([UCE,UTA,UO0A,VIM],X, [UCE,X,U0A,Y]) :-
es_digito(X),
es_digito(UTA),

Y is 10%VIM+X.
transicién_3([UCE,UTA,U0A,_VIM],X, [UCE,X,U0A,X]) :-

es_digito(X),

\+ es_digito(UTA).
transicién_3([UCE,_UTA,U0A,VIM],X, [Y,X,X,Y]) :-

\+ es_digito(X),

es_operaci6n(UOA),

T =.. [UOA,UCE,VIM],

Y is T.
transicién_3([_UCE,_UTA,=,VIM],X, [VIM,X,X,VIM]) :-

\+ es_digito(X).

En lo que sige usaremos la primera.

transicién(E1,A,E2) :-
transicién_1(E1,A,E2).

Ejercicio 7.9 Definir la relacién transiciones (+E1,+L,?E2) que se verifique si E2 es el esta-
do obtenido aplicando las acciones de la lista L al estado E1. Por ejemplo,

?7- estado_inicial(_E1), transiciones(_E1,[3,+,2,1,*,2,=],E2).
E2 = [48, =, =, 48]

Solucidon: La definicidn de transiciones es

transiciones(E, []1,E).

transiciones(E1, [X|L],E3) :-
transicién(E1,X,E2),
transiciones(E2,L,E3).

Ejercicio 7.10 Definir la relacién acciones(?L) que se verifique si L es una lista cuyos ele-
mentos son acciones. Por ejemplo,

?- acciones([2,+,3,7]).
Yes
?- acciones([2,+,37]).
No

Usarlo para calcular el niimero de posibles listas de acciones de longitud 3.

Solucion: La definicién de acciones es



7.2. Simulacion de una calculadora bésica 115

acciones([]).

acciones([XIL]) :-
accién(X),
acciones(L).

El ntimero N de listas de acciones de longitud 3 se calcula mediante la consulta

?7- findall(_L, (length(_L,3), acciones(_L)),_LAL3), length(_LAL3,N).
N = 3375

Ejercicio 7.11 Para realizar una operacion en la calculadora no todas las combinaciones de te-
clas (acciones) son vdlidas. Por ejemplo, no podemos teclear dos operaciones consecutivas o divi-
dir por cero. La siquiente relacion define las acciones vdlidas

acctones_validas(L) :-
acciones (L),
empieza_por_digito(L),
not(tiene_operaciones_consecutivas(L)),
not(tiene_resultado_intermedio (L)),
not(divide_por_cero(L)),
termina_en_digito_y_resultado(L).

En los apartados de este ejercicio se definen de las relaciones auxiliares.

1. Definir la relacion empieza_por_digito(?L) que se verifique si el primer elemento de la
lista L es un digito.

2. Definir la relacién tiene_operaciones_consecutivas(?L) que se verifique sila lista L
contiene dos operaciones consecutivas.

3. Definir la relacion tiene_resultado_intermedio(?L) que se verifique si la lista L con-
tiene el simbolo = en una posicion que no es la iiltima.

4. Definir la relacion divide_por_cero(?L) que se verifique si en la lista L aparecen de
manera consecutiva el simbolo / y un cero.

5. Definir la relacion termina_en_digito_y_resultado(?L) que se verifique si en la lista
L los iiltimos elementos son un digito y el simbolo =.

Solucidén:

empieza_por_digito([X|_L]) :-
es_digito(X).

tiene_operaciones_consecutivas(L) :-
append(_A, [X,Y|_Bl,L),
es_operacidon(X),
es_operacidén(Y).

tiene_resultado_intermedio(L) :-
append(_A, [=, _Y|_B],L).

divide_por_cero(L) :-
append(_A,[/,0|_B],L).

termina_en_digito_y_resultado(L) :-
reverse(L, [=,X]_]),
es_digito(X).

Ejercicio 7.12 Calcular el niimero de posibles listas de acciones vdilidas de longitud 3.

Solucién: El ntiimero N de listas de acciones vélidad de longitud 3 se calcula me-
diante la consulta

?- findall(_L, (length(_L,3), acciones_validas(_L)),_LAL3), length(_LAL3,N).

N = 100

Ejercicio 7.13 Definir la relacion calculo (+N,+M,-L) que se verifique si L es una lista de M
acciones vdlidas que aplicadas al estado inicial da como resultado el niimero N. Por ejemplo,

?- calculo(5,2,L).
L =[5, =] ;

No

?- calculo(5,3,L).
L = [0, 5, =] ;

No

?- calculo(5,4,L).
L = 1[0, 0, 5, =] ;
L =[0, +, 5, =] ;
L=1[1, +, 4, =] ;
L=1[1, %, 5, =] ;
L =1[2, +, 3, =]

Yes

Solucidén:

calculo(N,M,L) :-
estado_inicial(E1),
length(L,M),
acciones_validas(L),
transiciones(E1,L, [N,=,=,N]).

7.3. Problema de las subastas

Ejercicio 7.14 En una subasta se hacen distintas ofertas. Cada oferta incluye un lote de pro-
ductos y un precio por dicho lote. Las ofertas realizadas se representan mediante la relacién
oferta(0,L,P) que se verifica si O es una oferta por el lote L con un coste P. Por ejemplo,
oferta(a, [1,2,3],30) representa la oferta a en la que se puja por el lote compuesto por los
objetos 1, 2 y 3 por un valor de 30 euros.

Para la aceptacion de las ofertas se observan las siquientes reglas:

» No puede aceptar dos ofertas que contienen un mismo objeto en sus lotes.
m Se prefieren las ofertas de mayor ganancia.

Definir la relacion aceptada(-L) que se verifique si L es una lista de ofertas aceptadas. Por
ejemplo, si las ofertas realizadas se definen por

oferta(a,[1,2,3],30).
oferta(db,[1,2,3],20).
oferta(c, [4],20).
oferta(d, [2,4],20).
oferta(e, [1,2],20).

entonces,

?7- aceptada(L).
L = [a, c]

Solucién: La definicién de aceptada es

aceptada(L) :-
aceptable(L),
ganancia(L,G),
not ((aceptable(L1), ganancia(L1,G1), G1 > G)).

La relacion aceptable(?7L) se verifica si L es una lista de ofertas aceptable; es decir,
una lista de ofertas que no contienen objetos comunes en sus lotes. Por ejemplo, con la
definicién anterior de ofertas/3,

7- aceptable(L).
[a, c] ;

= [a] ;

= [b, c] ;

= [b] ;

[c, e] ;

= [c] ;

= [d] ;

= [e] ;

= [1 ;

[ e o o R o e e
1} 1}

=
o

aceptable(L) :-
lista_de_ofertas(L1),
subconjunto(L,L1),
es_aceptable(L).

La relacion lista_de_ofertas(-L) se verifica si L es la lista de todas las ofertas. Por
ejemplo, con la definicién anterior de ofertas/3,

?7- lista_de_ofertas(L).
L=1[a, b, ¢, d, el

lista_de_ofertas(L) :-
findall(0,oferta(0,_,_),L).

La relacién subconjunto(?L1,+L2) se verifica si L1 es un subconjunto de L2. Por
ejemplo,

7- subconjunto(L, [a,b,c]).

L = [a, b, c] ;
L = [a, b] ;
L = [a, c] ;
L = [a] ;

L =[b, c] ;
L = [b] ;

L = [c] ;
L=1[;

=
o

subconjunto([]1,[]).

subconjunto ([X|L1], [XIL2]) :-
subconjunto(L1,L2).

subconjunto (L1, [_|L2]) :-
subconjunto(L1,L2).

Larelacion es_aceptable (+L) se verifica si la lista de ofertas L es aceptable; es decir,
no contiene ofertas con objetos comunes en sus lotes. Por ejemplo, con la definicién
anterior de ofertas/3,

?7- es_aceptable([c,d]).
No
7- es_aceptable([c,e]).
Yes

es_aceptable(L) :-
not(es_inaceptable(L)).

La relacién es_inaceptable (+L) se verifica si L es una lista de ofertas inaceptable;
es decir, contiene ofertas con objetos comunes en sus lotes. Por ejemplo, con la definiciéon
anterior de ofertas/3,

?7- es_inaceptable([c,d]).
Yes
?7- es_inaceptable([c,el).
No

es_inaceptable(L) :-
member (01,L),
member (02,L),
01 \= 02,
oferta(01,L1,_),
oferta(02,L2,_),
se_solapan(L1,L2).

La relaciéon se_solapan(+L1,+L2) se verifica si L1 y L2 se solapan; es decir, tienen
elementos comunes. Por ejemplo,

7- se_solapan([a,b,c], [d,b,e]).
Yes

7- se_solapan([a,b,c],[d,e]).
No

se_solapan(L1,L2) :-
member (X,L1),
member (X,L2) .

La relacién ganancia(+L,-G) se verifica si la ganancia de la lista de ofertas L es G.
Por ejemplo, con la definiciéon anterior de ofertas/3,

?7- ganancia([a,c],G).
G = 50

ganancia([],0).
ganancia([0|L],G) :-
oferta(0,_,G1),
ganancia(L,G2),
G is G1+G2.

